Euro Summer School in Orthogonal Polynomials and Special Functions, Leuven, Belgium, August 12-17, 2002

Wolfram Koepf: Computer Algebra Algorithms for Orthogonal Polynomials and Special Functions

What is a Computer Algebra System about?

```
> 40!;
```

815915283247897734345611269596115894272000000000

```
> binomial(123,45);
```

896647319101861715891695497019268 4

```
> 40!/binomial(123,45);
```

$$\frac{259583501872662387403704332451840000000000}{28526840447291687613402857 3}$$

```
> evalf(Pi,100);
```

3.141592653589793238462643383279502884197169399375105820974944592307816406286208998628034825342117068

```
> p:=(x+y)^10-(x-y)^10;
```

$$p := (x+y)^{10} - (x-y)^{10}$$

```
> expand(p);
```

$$20\,x^9\,y + 240\,x^7\,y^3 + 504\,x^5\,y^5 + 240\,x^3\,y^7 + 20\,x\,y^9$$

```
> factor(p);
```

$$4\,x\,y\,(5\,x^4 + 10\,x^2\,y^2 + y^4)\,(x^4 + 10\,x^2\,y^2 + 5\,y^4)$$

```
> solve({x^2+y^2/2=1,-x^2+y^2+1=0},{x,y});
```

$$\{y=0, x=-1\}, \{x=1, y=0\}$$

```
> plots[implicitplot]({x^2+y^2/2=1,-x^2+y^2+1=0},x=-3..3,y=-3..3);
```




```
> plot3d(x^2-y^2,x=-1..1,y=-1..1);
```



```
>
```

Computation of Power Series

```
> series(exp(x),x);
```

$$1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+\mathrm{O}(x^6)$$

The following algorithm is from
Koepf, Wolfram: Power Series in Computer Algebra, Journal of Symbolic Computation 13, 1992, 581-603

```
> read "FPS.mpl";
```

Package Formal Power Series, Maple 7

Copyright 1995, Dominik Gruntz, University of Basel

Copyright 2002, Detlef Müller & Wolfram Koepf, University of Kassel

```
> FPS(exp(x),x);
```

$$\sum_{k=0}^{\infty}\frac{x^k}{k!}$$

```
> infolevel[FPS]:=5:
> FPS(exp(x),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   DE of degree   1   found.
FPS/FPS:   DE =
```

$$\mathrm{F}'(x)-\mathrm{F}(x)=0$$

```
FPS/FPS:   RE =
```

$$\mathrm{a}(k+1)=\frac{\mathrm{a}(k)}{k+1}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    1
FPS/hypergeomRE:   RE:
```

$$(k+1)\,\mathrm{a}(k+1)=\mathrm{a}(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    0
FPS/hypergeomRE:   a(0) =   1
```

$$\sum_{k=0}^{\infty}\frac{x^k}{k!}$$

```
> FPS(exp(x^2),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   DE of degree   1   found.
FPS/FPS:   DE =
```

$$\mathrm{F}'(x)-2\,x\,\mathrm{F}(x)=0$$

```
FPS/FPS:   RE =
```

$$\mathrm{a}(k+1)=\frac{2\,\mathrm{a}(k-1)}{k+1}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    2
FPS/hypergeomRE:   RE:
```

$$(k+2)\,\mathrm{a}(k+2)=2\,\mathrm{a}(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    -1
FPS/hypergeomRE:   a(0) =   1
```

$$\sum_{k=0}^{\infty} \frac{x^{(2k)}}{k!}$$

a Puiseux series

```
> FPS(exp(sqrt(x)),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   looking for DE of degree   2
FPS/FPS:   DE of degree   2   found.
FPS/FPS:   DE =
```

$$4\,x\,\mathrm{F}''(x) + 2\,\mathrm{F}'(x) - \mathrm{F}(x) = 0$$

```
FPS/FPS:   RE =
```

$$\mathrm{a}(k+1) = \frac{1}{2}\,\frac{\mathrm{a}(k)}{(k+1)\,(2\,k+1)}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    1
FPS/hypergeomRE:   RE:
```

$$2\,(k+1)\,(2\,k+1)\,\mathrm{a}(k+1) = \mathrm{a}(k)$$

```
FPS/hypergeomRE:   RE modified to    k = 1/2*k
FPS/hypergeomRE:   => f :=    exp(x)
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    2
FPS/hypergeomRE:   RE:
```

$$(k+2)\,(k+1)\,\mathrm{a}(k+2) = \mathrm{a}(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    0
FPS/hypergeomRE:   a(0) =   1
FPS/hypergeomRE:   a(1) =   1
```

$$\left(\sum_{k=0}^{\infty} \frac{x^k}{(2\,k)!}\right) + \left(\sum_{k=0}^{\infty} \frac{x^{(k+1/2)}}{(2\,k+1)!}\right)$$

```
> FPS(arcsin(x),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   looking for DE of degree   2
FPS/FPS:   DE of degree   2   found.
FPS/FPS:   DE =
```

$$(-1 + x^2)\,\mathrm{F}''(x) + x\,\mathrm{F}'(x) = 0$$

```
FPS/FPS:   RE =
```

$$\mathrm{a}(k+2) = \frac{k^2\,\mathrm{a}(k)}{(k+1)\,(k+2)}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    2
FPS/hypergeomRE:   RE:
```

$$-(k+1)\,(k+2)\,\mathrm{a}(k+2) = -k^2\,\mathrm{a}(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    0
FPS/hypergeomRE:   a(0) =   0
FPS/hypergeomRE:   a(2*j) = 0   for all j>0.
FPS/hypergeomRE:   a(1) =   1
```

$$\sum_{k=0}^{\infty} \frac{(2\,k)!\,4^{(-k)}\,x^{(2\,k+1)}}{(k!)^2\,(2\,k+1)}$$

```
> infolevel[FPS]:=0:
```

computation in steps

```
> f[0]:=arcsin(x);
```

$$f_0 := \arcsin(x)$$

```
> f[1]:=diff(f[0],x);
```

$$f_1 := \frac{1}{\sqrt{1-x^2}}$$

```
> normal(f[1]/f[0]);
```

$$\frac{1}{\sqrt{1-x^2}\,\arcsin(x)}$$

```
> f[2]:=diff(f[1],x);
```

$$f_2 := \frac{x}{(1-x^2)^{(3/2)}}$$

```
> ansatz:=sum(c[k]*f[k],k=0..2);
```

$$ansatz := c_0 \arcsin(x) + \frac{c_1}{\sqrt{1-x^2}} + \frac{c_2\, x}{(1-x^2)^{(3/2)}}$$

```
> normal(subs(c[0]=0,ansatz));
```

$$-\frac{-c_1 + c_1 x^2 - c_2 x}{(1-x^2)^{(3/2)}}$$

```
> sol:=solve(normal(subs(c[0]=0,ansatz)),{c[1],c[2]});
```

$$sol := \{c_2 = c_2,\ c_1 = \frac{c_2\, x}{-1+x^2}\}$$

```
> DE:=c[0]*F(x)+c[1]*diff(F(x),x)+c[2]*diff(F(x),x$2);
```

$$DE := c_0\, \mathrm{F}(x) + c_1 \left(\frac{d}{dx}\mathrm{F}(x)\right) + c_2 \left(\frac{d^2}{dx^2}\mathrm{F}(x)\right)$$

```
> collect(numer(normal(subs(sol,c[0]=0,DE/c[2]))),diff)=0;
```

$$x\left(\frac{d}{dx}\mathrm{F}(x)\right) + (-1+x^2)\left(\frac{d^2}{dx^2}\mathrm{F}(x)\right) = 0$$

procedures combining these steps

```
> DE:=HolonomicDE(arcsin(x),F(x));
```

$$DE := (x-1)(x+1)\left(\frac{d^2}{dx^2}\mathrm{F}(x)\right) + x\left(\frac{d}{dx}\mathrm{F}(x)\right) = 0$$

```
> dsolve(DE,F(x));
```

$$\mathrm{F}(x) = \_C1 + \ln(x + \sqrt{-1+x^2})\,\_C2$$

```
> RE:=SimpleRE(arcsin(x),x,a(k));
```

$$RE := k^2\, \mathrm{a}(k) - (k+1)(k+2)\, \mathrm{a}(k+2) = 0$$

```
> rsolve(RE,a(k));
```

$$\mathrm{rsolve}(k^2\, \mathrm{a}(k) - (k+1)(k+2)\, \mathrm{a}(k+2) = 0,\ \mathrm{a}(k))$$

some final examples: a Laurent series

```
> FPS(arcsin(x)^2/x^5,x);
```

$$\sum_{k=0}^{\infty} \frac{(k!)^2\, 4^k\, x^{(2k-3)}}{(1+2k)!\,(k+1)}$$

and an asymptotic series

```
> FPS((erf(x)-1)*exp(x^2),x=infinity);
```

$$-\frac{\displaystyle\sum_{k=0}^{\infty} \frac{(-1)^k\,(2k)!\,4^{(-k)}\left(\frac{1}{x}\right)^{(2k+1)}}{k!}}{\sqrt{\pi}}$$

special functions

```
> FPS(LaguerreL(n,x),x);
```

$$\sum_{k=0}^{\infty} \frac{\mathrm{pochhammer}(-n,k)\, x^k}{(k!)^2}$$

```
>
```

Computation of Holonomic Differential Equations

Exercise 1: Find a holonomic differential equation for f(x)=sin(x)*exp(x)

Solution:

```
> f[0]:=sin(x)*exp(x);
```

$$f_0 := \sin(x)\,\mathrm{e}^x$$

```
> f[1]:=diff(f[0],x);
```

$$f_1 := \cos(x)\,\mathrm{e}^x + \sin(x)\,\mathrm{e}^x$$

```
> normal(f[1]/f[0]);
```

$$\frac{\cos(x)+\sin(x)}{\sin(x)}$$

```
> f[2]:=diff(f[1],x);
```

$$f_2 := 2\cos(x)\,\mathrm{e}^x$$

```
> ansatz:=expand(sum(c[k]*f[k],k=0..2));
```

$$ansatz := c_0 \sin(x)\,\mathrm{e}^x + c_1 \cos(x)\,\mathrm{e}^x + c_1 \sin(x)\,\mathrm{e}^x + 2\,c_2 \cos(x)\,\mathrm{e}^x$$

```
> sol:=solve({c[0]+c[1]=0,c[1]+2*c[2]=0},{c[0],c[1],c[2]});
```

$$sol := \{c_1 = -2\,c_2,\ c_0 = 2\,c_2,\ c_2 = c_2\}$$

```
> DE:=c[0]*F(x)+c[1]*diff(F(x),x)+c[2]*diff(F(x),x$2);
```

$$DE := c_0\,\mathrm{F}(x) + c_1\left(\frac{d}{dx}\mathrm{F}(x)\right) + c_2\left(\frac{d^2}{dx^2}\mathrm{F}(x)\right)$$

```
> DE:=collect(numer(normal(subs(sol,DE/c[0]))),diff)=0;
```

$$DE := 2\,\mathrm{F}(x) - 2\left(\frac{d}{dx}\mathrm{F}(x)\right) + \left(\frac{d^2}{dx^2}\mathrm{F}(x)\right) = 0$$

```
> f:='f':
```

example from Olde Daalhuis' lecture:

```
> HolonomicDE(-(-z)^alpha*exp(-z*lambda)*GAMMA(1+alpha)*GAMMA(-alpha,-z*lambda),F(z));
```

$$\lambda \operatorname{F}(z) + (1 + z\lambda - \alpha)\left(\frac{d}{dz}\operatorname{F}(z)\right) + \left(\frac{d^2}{dz^2}\operatorname{F}(z)\right)z = 0$$

```
> HolonomicDE(-(-z)^alpha*exp(-z*lambda)*GAMMA(1+alpha)*GAMMA(-alpha,-z*lambda),F(lambda));
```

$$z(1+\alpha)\operatorname{F}(\lambda) + (1 + z\lambda + \alpha)\left(\frac{d}{d\lambda}\operatorname{F}(\lambda)\right) + \left(\frac{d^2}{d\lambda^2}\operatorname{F}(\lambda)\right)\lambda = 0$$

```
>
```

Algebra of Holonomic Functions

```
> read "FPS.mpl";
```

Package Formal Power Series, Maple 7

Copyright 1995, Dominik Gruntz, University of Basel

Copyright 2002, Detlef Müller & Wolfram Koepf, University of Kassel

```
> with(share): with(gfun);
See ?share and ?share,contents for information about the share library
```

[*Laplace*, *algebraicsubs*, *algeqtodiffeq*, *algeqtoseries*, *algfuntoalgeq*, *borel*, *cauchyproduct*, *diffeq*diffeq*, *diffeq+diffeq*, *diffeqtohomdiffeq*, *diffeqtorec*, *guesseqn*, *guessgf*, *hadamardproduct*, *holexprtodiffeq*, *invborel*, *listtoalgeq*, *listtodiffeq*, *listtohypergeom*, *listtolist*, *listtoratpoly*, *listtorec*, *listtoseries*, *listtoseries/Laplace*, *listtoseries/egf*, *listtoseries/lgdegf*, *listtoseries/lgdogf*, *listtoseries/ogf*, *listtoseries/revegf*, *listtoseries/revogf*, *maxdegcoeff*, *maxdegeqn*, *maxordereqn*, *mindegcoeff*, *mindegeqn*, *minordereqn*, *optionsgf*, *poltodiffeq*, *poltorec*, *ratpolytocoeff*, *rec*rec*, *rec+rec*, *rectodiffeq*, *rectohomrec*, *rectoproc*, *seriestoalgeq*, *seriestodiffeq*, *seriestohypergeom*, *seriestolist*, *seriestoratpoly*, *seriestorec*, *seriestoseries*]

The function sin(x)*exp(x), again:

```
> DE1:=diff(F(x),x$2)+F(x)=0;
```

$$DE1 := \left(\frac{d^2}{dx^2}\operatorname{F}(x)\right) + \operatorname{F}(x) = 0$$

```
> DE2:=diff(F(x),x)-F(x)=0;
```

$$DE2 := \left(\frac{d}{dx}\operatorname{F}(x)\right) - \operatorname{F}(x) = 0$$

```
> `diffeq*diffeq`(DE1,DE2,F(x));
```

$$2\operatorname{F}(x) - 2\left(\frac{d}{dx}\operatorname{F}(x)\right) + \left(\frac{d^2}{dx^2}\operatorname{F}(x)\right)$$

and the sum sin(x)+exp(x) satisfies

```
> `diffeq+diffeq`(DE1,DE2,F(x));
```

$$\left(\frac{d^3}{dx^3}F(x)\right)+\left(\frac{d}{dx}F(x)\right)-\left(\frac{d^2}{dx^2}F(x)\right)-F(x)$$

Now a more complicated example: exp(x)*Ai(x)

```
> DE1:=diff(F(x),x)-F(x)=0;
```

$$DE1 := \left(\frac{d}{dx}F(x)\right)-F(x)=0$$

```
> DE2:=SimpleDE(AiryAi(x),F(x));
```

$$DE2 := \left(\frac{d^2}{dx^2}F(x)\right)-x\,F(x)=0$$

```
> `diffeq*diffeq`(DE1,DE2,F(x));
```

$$(-x+1)\,F(x)+\left(\frac{d^2}{dx^2}F(x)\right)-2\left(\frac{d}{dx}F(x)\right)$$

```
> `diffeq+diffeq`(DE1,DE2,F(x));
```

$$\{(D^{(2)})(F)(0)=\_C_0,$$

$$(-x+1+x^2)\,F(x)+(x-x^2)\left(\frac{d}{dx}F(x)\right)-x\left(\frac{d^2}{dx^2}F(x)\right)+(x-1)\left(\frac{d^3}{dx^3}F(x)\right)\}$$

Similar algorithms exist for sequences an recurrence equations. Assume we want to find a recurrence equation w.r.t. k for

```
> binomial(n,k)+binomial(k,n);
```

$$\mathrm{binomial}(n,k)+\mathrm{binomial}(k,n)$$

The binomial coefficient binomial(n,k) (first summand) satisfies the equation

```
> S(k+1)/S(k)=expand(binomial(n,k+1)/binomial(n,k));
```

$$\frac{S(k+1)}{S(k)}=\frac{n-k}{k+1}$$

w.r.t. k. This gives the holonomic recurrence equation

```
> RE1:=collect(numer(normal(S(k+1)-expand(binomial(n,k+1)/binom
  ial(n,k))*S(k))),S,factor);
```

$$RE1 := (k+1)\,S(k+1)+(k-n)\,S(k)$$

The binomial coefficient binomial(k,n) (second summand) satisfies the equation

```
> S(k+1)/S(k)=expand(binomial(k+1,n)/binomial(k,n));
```

$$\frac{S(k+1)}{S(k)}=\frac{k+1}{k+1-n}$$

w.r.t. k. This gives the holonomic recurrence equation

```
> RE2:=collect(numer(normal(S(k+1)-expand(binomial(k+1,n)/binom
  ial(k,n))*S(k))),S,factor);
```

$$RE2 := (n-k-1)\,S(k+1)+(k+1)\,S(k)$$

Therefore we get for the sum

```
> `rec+rec`(RE1,RE2,S(k));
```

$$\{(3k^2n^2+3n^2+2k^4+6kn^2-n^3k-6n+9k^3+13k^2+6k-14k^2n-4k^3n-16nk$$
$$-n^3)\,S(k)+(4-6k^2n^2+4k^3n+12k^2n-12kn^2+10nk-n^4+4n^3-5n^2+2n$$

$$+10k+2k^3+8k^2+4n^3k)\,S(k+1)+(n^3k+2n^3-3k^2n^2-9kn^2-6n^2+4k^3n$$
$$+16k^2n+19nk+6n-2k^4-11k^3-21k^2-16k-4)\,S(k+2),\ S(1)=n\_C_0+\_C_1,$$
$$S(0)=\_C_0-\_C_1 n+\_C_1\}$$

Just for fun we compute the recurrence equation for the product

```
> `rec*rec`(RE1,RE2,S(k));
```

$$(n-k)\,S(k)+(n-k-1)\,S(k+1)$$

```
>
```

Identification of Hypergeometric Functions

We are interested in

```
> s:=Sum(binomial(n,k)^2,k=0..infinity);
```

$$s:=\sum_{k=0}^{\infty}\mathrm{binomial}(n,k)^2$$

```
> F:=k->binomial(n,k)^2;
```

$$F:=k\to\mathrm{binomial}(n,k)^2$$

```
> r:=F(k+1)/F(k);
```

$$r:=\frac{\mathrm{binomial}(n,k+1)^2}{\mathrm{binomial}(n,k)^2}$$

```
> expand(r);
```

$$\frac{(n-k)^2}{(k+1)^2}$$

Hence

```
> s=hypergeom([-n,-n],[1],1);
```

$$\sum_{k=0}^{\infty}\mathrm{binomial}(n,k)^2=\mathrm{hypergeom}([-n,-n],[1],1)$$

Check

```
> convert(s,hypergeom);
```

$$\frac{\Gamma(2n+1)}{\Gamma(1+n)^2}$$

Maple simplifies completely, hence we don't see the hypergeometric form. The same applies to

```
> simplify(hypergeom([-n,-n],[1],1));
```

$$\frac{\Gamma(2n+1)}{\Gamma(1+n)^2}$$

This gives the hypergeometric form:

```
> sumtools[Sumtohyper](F(k),k);
```

$$\mathrm{Hypergeom}([-n,-n],[1],1)$$

Another example

```
> F:=binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k;
```

$$F := \text{binomial}(n, k)\,\text{binomial}(-n-1, k)\left(\frac{1}{2}-\frac{x}{2}\right)^k$$

```
> Sum(F,k=0..n)=sumtools[Sumtohyper](F,k);
```

$$\sum_{k=0}^{n} \text{binomial}(n, k)\,\text{binomial}(-n-1, k)\left(\frac{1}{2}-\frac{x}{2}\right)^k = \text{Hypergeom}\left([-n, 1+n], [1], \frac{1}{2}-\frac{x}{2}\right)$$

```
> with(sumtools);
```

[*Hypersum*, *Sumtohyper*, *extended_gosper*, *gosper*, *hyperrecursion*, *hypersum*, *hyperterm*, *simpcomb*, *sumrecursion*, *sumtohyper*]

```
> ?sumtools;
> interface(verboseproc=2);
> print(sumtools[Sumtohyper]);
```

proc(f, k) ... **end proc**

Details of this algorithm and an implementation can be found in the book
Wolfram Koepf: *Hypergeometric Summation*, Vieweg, Braunschweig/Wiesbaden, 1998

We can combine the FPS and the identification algorithm:

```
> s:=FPS(exp(x),x,k);
```

$$s := \sum_{k=0}^{\infty} \frac{x^k}{k!}$$

```
> op(1,s);
```

$$\frac{x^k}{k!}$$

```
> Sumtohyper(op(1,s),k);
```

$$\text{Hypergeom}([\ ], [\ ], x)$$

```
> s:='s':
```

Exercise 3: Write cos(x) in hypergeometric notation.

Solution:

```
> Sumtohyper((-1)^k/(2*k)!*x^(2*k),k);
```

$$\text{Hypergeom}\left([\ ], \left[\frac{1}{2}\right], -\frac{x^2}{4}\right)$$

or, combining FPS and hsum:

```
> fps:=FPS(cos(x),x,k);
```

$$fps := \sum_{k=0}^{\infty} \frac{(-1)^k x^{(2k)}}{(2k)!}$$

```
> Sumtohyper(op(1,fps),k);
```

$$\text{Hypergeom}\left([\ ], \left[\frac{1}{2}\right], -\frac{x^2}{4}\right)$$

```
>
```

Computation of Recurrence Equations for Hypergeometric

Functions

How does one generate the result

```
> Sum(binomial(n,k),k=0..n)=sum(binomial(n,k),k=0..n);
```

$$\sum_{k=0}^{n} \mathrm{binomial}(n, k) = 2^n$$

We do the following more complicated example with Maple:

```
> Sum(k*binomial(n,k),k=0..n)=sum(k*binomial(n,k),k=0..n);
```

$$\sum_{k=0}^{n} k\,\mathrm{binomial}(n, k) = \frac{2^n\, n}{2}$$

```
> F:=(n,k)->k*binomial(n,k);
```

$$F := (n, k) \to k\,\mathrm{binomial}(n, k)$$

```
> ansatz:=sum(sum(a(j,i)*F(n+j,k+i),i=0..1),j=0..1);
```

$$ansatz := \mathrm{a}(0, 0)\, k\,\mathrm{binomial}(n, k) + \mathrm{a}(0, 1)\,(k+1)\,\mathrm{binomial}(n, k+1) + \mathrm{a}(1, 0)\, k\,\mathrm{binomial}(1+n, k) + \mathrm{a}(1, 1)\,(k+1)\,\mathrm{binomial}(1+n, k+1)$$

```
> ansatz:=ansatz/F(n,k);
```

$$ansatz := (\mathrm{a}(0, 0)\, k\,\mathrm{binomial}(n, k) + \mathrm{a}(0, 1)\,(k+1)\,\mathrm{binomial}(n, k+1) + \mathrm{a}(1, 0)\, k\,\mathrm{binomial}(1+n, k) + \mathrm{a}(1, 1)\,(k+1)\,\mathrm{binomial}(1+n, k+1)) / (k\,\mathrm{binomial}(n, k))$$

```
> ansatz:=expand(ansatz);
```

$$ansatz := \mathrm{a}(0, 0) + \frac{\mathrm{a}(0, 1)\, n}{k+1} - \frac{k\,\mathrm{a}(0, 1)}{k+1} + \frac{\mathrm{a}(0, 1)\, n}{k\,(k+1)} - \frac{\mathrm{a}(0, 1)}{k+1} + \frac{\mathrm{a}(1, 0)}{n-k+1} + \frac{\mathrm{a}(1, 0)\, n}{n-k+1} + \frac{\mathrm{a}(1, 1)}{k+1} + \frac{\mathrm{a}(1, 1)\, n}{k+1} + \frac{\mathrm{a}(1, 1)}{k\,(k+1)} + \frac{\mathrm{a}(1, 1)\, n}{k\,(k+1)}$$

```
> ansatz:=normal(ansatz);
```

$$ansatz := (-k^2\,\mathrm{a}(0, 0) + k^2\,\mathrm{a}(0, 1) + \mathrm{a}(0, 0)\, k\, n + \mathrm{a}(1, 0)\, n\, k - \mathrm{a}(1, 1)\, k\, n - \mathrm{a}(1, 1)\, k + \mathrm{a}(1, 0)\, k - 2\,\mathrm{a}(0, 1)\, n\, k + \mathrm{a}(0, 0)\, k - k\,\mathrm{a}(0, 1) + \mathrm{a}(0, 1)\, n^2 + 2\,\mathrm{a}(1, 1)\, n + \mathrm{a}(0, 1)\, n + \mathrm{a}(1, 1) + \mathrm{a}(1, 1)\, n^2) / ((n-k+1)\, k)$$

```
> ansatz:=numer(ansatz);
```

$$ansatz := -k^2\,\mathrm{a}(0, 0) + k^2\,\mathrm{a}(0, 1) + \mathrm{a}(0, 0)\, k\, n + \mathrm{a}(1, 0)\, n\, k - \mathrm{a}(1, 1)\, k\, n - \mathrm{a}(1, 1)\, k + \mathrm{a}(1, 0)\, k - 2\,\mathrm{a}(0, 1)\, n\, k + \mathrm{a}(0, 0)\, k - k\,\mathrm{a}(0, 1) + \mathrm{a}(0, 1)\, n^2 + 2\,\mathrm{a}(1, 1)\, n + \mathrm{a}(0, 1)\, n + \mathrm{a}(1, 1) + \mathrm{a}(1, 1)\, n^2$$

```
> eqs:={coeffs(ansatz,k)};
```

$$eqs := \{\mathrm{a}(1, 1) + \mathrm{a}(1, 1)\, n^2 + \mathrm{a}(0, 1)\, n^2 + 2\,\mathrm{a}(1, 1)\, n + \mathrm{a}(0, 1)\, n, -\mathrm{a}(0, 0) + \mathrm{a}(0, 1), \mathrm{a}(1, 0)\, n - \mathrm{a}(1, 1)\, n - \mathrm{a}(1, 1) + \mathrm{a}(0, 0)\, n - 2\,\mathrm{a}(0, 1)\, n + \mathrm{a}(0, 0) - \mathrm{a}(0, 1) + \mathrm{a}(1, 0)\}$$

```
> sol:=solve(eqs,{seq(seq(a(j,i),j=0..1),i=0..1)});
```

$$sol := \left\{\mathrm{a}(0, 1) = -\frac{(1+n)\,\mathrm{a}(1, 1)}{n}, \mathrm{a}(1, 0) = 0, \mathrm{a}(0, 0) = -\frac{(1+n)\,\mathrm{a}(1, 1)}{n}, \mathrm{a}(1, 1) = \mathrm{a}(1, 1)\right\}$$

```
> re:=sum(sum(a(j,i)*f(n+j,k+i),i=0..1),j=0..1);
```

$$re := a(0,0)\,f(n,k) + a(0,1)\,f(n,k+1) + a(1,0)\,f(1+n,k) + a(1,1)\,f(1+n,k+1)$$

```
> re:=subs(sol,re);
```

$$re := -\frac{(1+n)\,a(1,1)\,f(n,k)}{n} - \frac{(1+n)\,a(1,1)\,f(n,k+1)}{n} + a(1,1)\,f(1+n,k+1)$$

```
> re:=numer(normal(re/a(1,1)));
```

$$re := -f(n,k) - f(n,k)\,n - f(n,k+1) - f(n,k+1)\,n + f(1+n,k+1)\,n$$

```
> RE:=subs({seq(seq(f(n+j,k+i)=s(n+j),i=0..1),j=0..1)},re)=0;
```

$$RE := -2\,s(n) - 2\,s(n)\,n + s(1+n)\,n = 0$$

Now we use the implementation from the book
Wolfram Koepf: *Hypergeometric Summation*, Vieweg, Braunschweig/Wiesbaden, 1998

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

```
> libname:=libname,"C:/Dokumente und Einstellungen/koepf/Eigene
  Dateien/Koepf/Vorträge/SummerSchool/hsum";
```

libname := "C:\Programme\Maple 8/lib", "C:/Dokumente und Einstellungen/koepf/Eigene \
Dateien/Koepf/Vorträge/SummerSchool/hsum"

```
> ?hsum
> fasenmyer(k*binomial(n,k),k,s(n),1,1);
```

$$s(1+n)\,n - 2\,s(n)\,(1+n) = 0$$

```
> fasenmyer(binomial(n,k)^2,k,s(n),1,1);
Error, (in kfreerec) No kfree recurrence equation of order (1,1) exists
```

```
> fasenmyer(binomial(n,k)^2,k,s(n),2,1);
```

$$(2+n)\,s(2+n) - 2\,s(1+n)\,(2\,n+3) = 0$$

```
> fasenmyer(binomial(n-k,k),k,s(n),2,1);
```

$$s(2+n) - s(n) - s(1+n) = 0$$

```
> [seq(sum(binomial(n-k,k),k=0..n),n=0..10)]; n:='n':
```

$$[1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89]$$

```
> fasenmyer((-1)^k*binomial(n,k)^2,k,s(n),2,2);
```

$$(2+n)\,s(2+n) + 4\,s(n)\,(1+n) = 0$$

```
> fasenmyer(binomial(n,k)^3,k,s(n),2,1);
Error, (in kfreerec) No kfree recurrence equation of order (2,2) exists
```

```
> fasenmyer(binomial(n,k)^3,k,s(n),3,1);
```

$$(3\,n+4)\,(3+n)^2\,s(3+n) - 2\,(9\,n^3 + 57\,n^2 + 116\,n + 74)\,s(2+n)$$
$$-(3\,n+5)\,(15\,n^2 + 55\,n + 48)\,s(1+n) - 8\,(3\,n+7)\,(1+n)^2\,s(n) = 0$$

Legendre polynomials

```
> Sum(binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k,k=0..n);
```

$$\sum_{k=0}^{n} \mathrm{binomial}(n,k)\,\mathrm{binomial}(-n-1,k)\left(\frac{1}{2}-\frac{x}{2}\right)^k$$

This corresponds to the hypergeometric representation

```
> Sumtohyper(binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k,k);
```

$$\mathrm{Hypergeom}\left([1+n, -n], [1], \frac{1}{2} - \frac{x}{2}\right)$$

```
> fasenmyer(binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k,k,s(n),2
,1);
```

$$(2+n)\,\mathrm{s}(2+n) - x\,(2\,n+3)\,\mathrm{s}(1+n) + (1+n)\,\mathrm{s}(n) = 0$$

Exercise 5: Compute a three-term recurrence equation for the Laguerre polynomials.

Solution

The Laguerre polynomials have the hypergeometric representation

```
> LaguerreL(n,x)=hypergeom([-n],[1],x);
```

$$\mathrm{LaguerreL}(n, x) = \mathrm{hypergeom}([-n], [1], x)$$

Therefore we get

```
> fasenmyer(hyperterm([-n],[1],x,k),k,s(n),2,1);
```

$$(2+n)\,\mathrm{s}(2+n) - (2\,n - x + 3)\,\mathrm{s}(1+n) + (1+n)\,\mathrm{s}(n) = 0$$

The generalized Laguerre polynomials have the hypergeometric representation

```
> LaguerreL(n,alpha,x)=binomial(n+alpha,n)*hypergeom([-n],[1+al
pha],x);
```

$$\mathrm{LaguerreL}(n, \alpha, x) = \mathrm{binomial}(n+\alpha, n)\,\mathrm{hypergeom}([-n], [1+\alpha], x)$$

Therefore we get

```
> fasenmyer(binomial(n+alpha,n)*hyperterm([-n],[1+alpha],x,k),k
,s(n),2,1);
```

$$(2+n)\,\mathrm{s}(2+n) - (2\,n+\alpha+3-x)\,\mathrm{s}(1+n) + (1+n+\alpha)\,\mathrm{s}(n) = 0$$

```
>
```

Indefinite Summation

Indefinite sum of k*k!

```
> s:=sum(k*k!,k);
```

$$s := k!$$

Check:

```
> difference:=subs(k=k+1,s)-s;
```

$$\mathit{difference} := (k+1)! - k!$$

```
> simplify(difference);
```

$$k\,\Gamma(k+1)$$

```
> simplify(difference-k*k!);
```

$$0$$

```
> infolevel[sum]:=5:
> sum((-1)^k*binomial(n,k),k);
```

```
sum/indefnew:   indefinite summation
sum/extgosper:   applying Gosper algorithm to a(   k   ):=   (-1)^k*binomial
(n,k)
sum/gospernew/internal:   a(   k   )/a(   k   -1):=   -(n-k+1)/k
sum/gospernew/internal:   Gosper's algorithm applicable
sum/gospernew/internal:   p:=   1
sum/gospernew/internal:   q:=   -n-1+k
```

```
sum/gospernew/internal:   r:=   k
sum/gospernew/internal:   degreebound:=   0
sum/gospernew/internal:   solving equations to find f
sum/gospernew/internal:   Gosper's algorithm successful
sum/gospernew/internal:   f:=   -1/n
sum/indefnew:   indefinite summation finished
```

$$-\frac{k\,(-1)^k\,\mathrm{binomial}(n,k)}{n}$$

```
> with(sumtools);
Warning, these names have been redefined: Sumtohyper, extended_gosper,
gosper, hyperterm, simpcomb, sumrecursion
```

$$[Hypersum, Sumtohyper, extended\_gosper, gosper, hyperrecursion, hypersum, hyperterm, simpcomb, sumrecursion, sumtohyper]$$

```
> gosper((-1)^k*binomial(n,k),k);
sum/gospernew/internal:   a(   k   )/a(   k   -1):=   -(n-k+1)/k
sum/gospernew/internal:   Gosper's algorithm applicable
sum/gospernew/internal:   p:=   1
sum/gospernew/internal:   q:=   -n-1+k
sum/gospernew/internal:   r:=   k
sum/gospernew/internal:   degreebound:=   0
sum/gospernew/internal:   solving equations to find f
sum/gospernew/internal:   Gosper's algorithm successful
sum/gospernew/internal:   f:=   -1/n
```

$$-\frac{k\,(-1)^k\,\mathrm{binomial}(n,k)}{n}$$

Example from SIAM Reviews 36, 1994, Problem 94-2

```
> Sum((-1)^(k+1)*(4*k+1)*(2*k)!/(k!*4^k*(2*k-1)*(k+1)!),k=1..infinity);
```

$$\sum_{k=1}^{\infty}\frac{(-1)^{(k+1)}\,(4\,k+1)\,(2\,k)!}{k!\,4^k\,(2\,k-1)\,(k+1)!}$$

```
> sum((-1)^(k+1)*(4*k+1)*(2*k)!/(k!*4^k*(2*k-1)*(k+1)!),k);
sum/indefnew:   indefinite summation
sum/extgosper:   applying Gosper algorithm to a(   k   ):=   (-1)^(k+1)*(4*k
+1)*(2*k)!/k!/(4^k)/(2*k-1)/(k+1)!
sum/gospernew/internal:   a(   k   )/a(   k   -1):=   -1/2*(4*k+1)/(4*k-3)/
(k+1)*(2*k-3)
sum/gospernew/internal:   Gosper's algorithm applicable
sum/gospernew/internal:   p:=   4*k+1
sum/gospernew/internal:   q:=   -2*k+3
sum/gospernew/internal:   r:=   2*k+2
sum/gospernew/internal:   degreebound:=   0
sum/gospernew/internal:   solving equations to find f
sum/gospernew/internal:   Gosper's algorithm successful
sum/gospernew/internal:   f:=   -1
sum/indefnew:   indefinite summation finished
```

$$-\frac{2\,(k+1)\,(-1)^{(k+1)}\,(2\,k)!}{k!\,4^k\,(2\,k-1)\,(k+1)!}$$

```
> sum((-1)^(k+1)*(4*k+1)*(2*k)!/(k!*4^k*(2*k-1)*(k+1)!),k=1..infinity);
sum/infinite:   infinite summation
```

$$1$$

```
> infolevel[sum]:=0:
```

We do a more complicated example

```
> s:=k!*binomial(n,k)/(n-k);
```

$$s := \frac{k!\ \mathrm{binomial}(n, k)}{n-k}$$

```
> a:=subs(k=k+3,s)-s;
```

$$a := \frac{(k+3)!\ \mathrm{binomial}(n, k+3)}{n-k-3} - \frac{k!\ \mathrm{binomial}(n, k)}{n-k}$$

```
> b:=gosper(a,k);
```

$$b := (n-k-3)\,(2-n^3-4\,n^3\,k+6\,k^2\,n^2-4\,k^3\,n-3\,k^2\,n+3\,k\,n^2+k^4+k^3-3\,n+3\,k+n^4)\left(\frac{(k+3)!\ \mathrm{binomial}(n, k+3)}{n-k-3} - \frac{k!\ \mathrm{binomial}(n, k)}{n-k}\right) \Big/ ((n^4-4\,n^3\,k-3\,n^3+6\,k^2\,n^2+9\,k\,n^2+2\,n^2-4\,k^3\,n-9\,k^2\,n-4\,n\,k+k^4+3\,k^3+2\,k^2-n+k+3)\,(n-k-2)\,(n-k-1))$$

```
> gosper(b,k);
```

$$\mathit{FAIL}$$

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

```
> gosper(b,k);
Error, (in gosper) No hypergeometric term antidifference exists
```

```
> a:='a': b:='b':
>
```

Gosper's Algorithm in Detail

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

first example

```
> a:=k*k!;
```

$$a := k\ k!$$

```
> rat:=subs(k=k+1,a)/a;
```

$$rat := \frac{(k+1)\,(k+1)!}{k\ k!}$$

```
> rat:=normal(expand(rat));
```

$$rat := \frac{k^2+2\,k+1}{k}$$

```
> q:=numer(rat);
```

$$q := k^2 + 2k + 1$$

```
> r:=denom(rat);
```

$$r := k$$

```
> p:=1;
```

$$p := 1$$

q(k) and r(k+j) have a nontrivial gcd for j=1:

```
> gcd(q,subs(k=k+1,r));
```

$$k + 1$$

```
> pqr:=update(p,subs(k=k-1,q),subs(k=k-1,r),k);
```

$$pqr := [k, k, 1]$$

```
> p:=op(1,pqr); q:=op(2,pqr); r:=op(3,pqr);
```

$$p := k$$

$$q := k$$

$$r := 1$$

```
> f:='f':
> RE:=subs(k=k+1,q)*f(k)-subs(k=k+1,r)*f(k-1)=p;
```

$$RE := (k + 1)\, f(k) - f(k - 1) = k$$

```
> rsolve(RE,f(k));
```

$$1 + \frac{-1 + f(0)}{\Gamma(k + 2)}$$

```
> f:=findf(p,q,r,k);
```

$$f := 1$$

```
> s:=r/p*subs(k=k-1,f)*a;
```

$$s := k!$$

second example

```
> a:=(-1)^k*binomial(n,k);
```

$$a := (-1)^k \operatorname{binomial}(n, k)$$

```
> rat:=subs(k=k+1,a)/a;
```

$$rat := \frac{(-1)^{(k+1)} \operatorname{binomial}(n, k + 1)}{(-1)^k \operatorname{binomial}(n, k)}$$

```
> rat:=normal(expand(rat));
```

$$rat := -\frac{n - k}{k + 1}$$

```
> q:=numer(rat);
```

$$q := k - n$$

```
> r:=denom(rat);
```

$$r := k + 1$$

```
> p:=1;
```

$$p := 1$$

q(k) and r(k+j) have no nontrivial gcd for n a symbol, but for negative integer n. We will come back to this case later.

```
> pqr:=update(p,subs(k=k-1,q),subs(k=k-1,r),k);
```

$$pqr := [1, -n-1+k, k]$$

```
> p:=op(1,pqr); q:=op(2,pqr); r:=op(3,pqr);
```

$$p := 1$$

$$q := -n-1+k$$

$$r := k$$

```
> f:='f':
> RE:=subs(k=k+1,q)*f(k)-subs(k=k+1,r)*f(k-1)=p;
```

$$RE := (k-n)\,\mathrm{f}(k) - (k+1)\,\mathrm{f}(k-1) = 1$$

```
> sol:=rsolve(RE,f(k));
```

$$sol := -\frac{1}{1+n} + \frac{(\mathrm{f}(0)\,n+1+\mathrm{f}(0))\,\Gamma(k+2)\,\Gamma(-n+1)}{(1+n)\,\Gamma(-n+k+1)}$$

```
> f:=findf(p,q,r,k);
```

$$f := -\frac{1}{n}$$

```
> s:=r/p*subs(k=k-1,f)*a;
```

$$s := -\frac{k\,(-1)^k\,\mathrm{binomial}(n, k)}{n}$$

Now we consider the particular case n=-10.

```
> a:=(-1)^k*binomial(-10,k);
```

$$a := (-1)^k\,\mathrm{binomial}(-10, k)$$

```
> rat:=subs(k=k+1,a)/a;
```

$$rat := \frac{(-1)^{(k+1)}\,\mathrm{binomial}(-10, k+1)}{(-1)^k\,\mathrm{binomial}(-10, k)}$$

```
> rat:=normal(expand(rat));
```

$$rat := \frac{10+k}{k+1}$$

```
> q:=numer(rat);
```

$$q := 10+k$$

```
> r:=denom(rat);
```

$$r := k+1$$

```
> p:=1;
```

$$p := 1$$

q(k) and r(k+j) have a nontrivial gcd for j=9:

```
> gcd(q,subs(k=k+9,r));
```

$$10+k$$

```
> pqr:=update(p,subs(k=k-1,q),subs(k=k-1,r),k);
```

$$pqr := [(k+9)(k+8)(k+7)(k+6)(k+5)(k+4)(k+3)(k+2)(k+1), 1, 1]$$

```
> p:=op(1,pqr); q:=op(2,pqr); r:=op(3,pqr);
```

$$p := (k+9)(k+8)(k+7)(k+6)(k+5)(k+4)(k+3)(k+2)(k+1)$$

$$q := 1$$

$$r := 1$$

```
> f:='f':
> RE:=subs(k=k+1,q)*f(k)-subs(k=k+1,r)*f(k-1)=p;
```

$$RE := \mathrm{f}(k) - \mathrm{f}(k-1) = (k+9)(k+8)(k+7)(k+6)(k+5)(k+4)(k+3)(k+2)(k+1)$$

```
> sol:=rsolve(RE,f(k));
```

$$sol := \mathrm{f}(0) - 362880 + 362880\,(k+1)\left(\frac{k}{2}+1\right)\left(\frac{k}{3}+1\right)\left(\frac{k}{4}+1\right)\left(\frac{k}{5}+1\right)\left(\frac{k}{6}+1\right)\left(\frac{k}{7}+1\right)\left(\frac{k}{8}+1\right)\left(\frac{k}{9}+1\right)\left(1+\frac{k}{10}\right)$$

```
> f:=findf(p,q,r,k);
```

$$f := 1062864\,k + \frac{6376788}{5}k^2 + 840950\,k^3 + 341693\,k^4 + \frac{180411}{2}k^5 + \frac{157773}{10}k^6 + 1815\,k^7 + 132\,k^8 + \frac{11}{2}k^9 + \frac{1}{10}k^{10}$$

```
> specials:=r/p*subs(k=k-1,f)*a;
```

$$specials := \left(1062864\,k - 1062864 + \frac{6376788\,(k-1)^2}{5} + 840950\,(k-1)^3 + 341693\,(k-1)^4 + \frac{180411\,(k-1)^5}{2} + \frac{157773\,(k-1)^6}{10} + 1815\,(k-1)^7 + 132\,(k-1)^8 + \frac{11\,(k-1)^9}{2} + \frac{(k-1)^{10}}{10}\right)(-1)^k\,\mathrm{binomial}(-10,k)\,/\,((k+9)(k+8)(k+7)(k+6)(k+5)(k+4)(k+3)(k+2)(k+1))$$

```
> difference:=simplify(specials-subs(n=-10,s));
```

$$difference := \frac{362880\,(-1)^{(k+1)}\,\mathrm{binomial}(-10,k)}{(k+9)(k+8)(k+7)(k+6)(k+5)(k+4)(k+3)(k+2)(k+1)}$$

```
> simplify(difference);
```

$$\frac{362880\,(-1)^{(k+1)}\,\mathrm{binomial}(-10,k)}{(k+9)(k+8)(k+7)(k+6)(k+5)(k+4)(k+3)(k+2)(k+1)}$$

```
> [seq(difference,k=1..10)]; k:='k':
```

$$[-1, -1, -1, -1, -1, -1, -1, -1, -1, -1]$$

third example

```
> a:=binomial(n,k);
```

$$a := \mathrm{binomial}(n,k)$$

```
> rat:=subs(k=k+1,a)/a;
```

$$rat := \frac{\mathrm{binomial}(n,k+1)}{\mathrm{binomial}(n,k)}$$

```
> rat:=normal(expand(rat));
```

$$rat := \frac{n-k}{k+1}$$

```
> q:=numer(rat);
```

$$q := n - k$$

```
> r:=denom(rat);
```

$$r := k + 1$$

```
> p:=1;
```

$$p := 1$$

```
> pqr:=update(p,subs(k=k-1,q),subs(k=k-1,r),k);
```

$$pqr := [1, n-k+1, k]$$

```
> p:=op(1,pqr); q:=op(2,pqr); r:=op(3,pqr);
```

$$p := 1$$

$$q := n - k + 1$$

$$r := k$$

```
> f:='f':
> RE:=subs(k=k+1,q)*f(k)-subs(k=k+1,r)*f(k-1)=p;
```

$$RE := (n-k)\,\mathrm{f}(k) - (k+1)\,\mathrm{f}(k-1) = 1$$

```
> rsolve(RE,f(k));
```

$$-\frac{\mathrm{hypergeom}([1, -n+k+1], [k+3], -1)}{k+2} + \frac{(-n-2+2^{(1+n)}+\mathrm{f}(0)\,n+\mathrm{f}(0)\,n^2)\,(-1)^k\,\Gamma(-n-1)\,\Gamma(k+2)}{\Gamma(-n+k+1)}$$

```
> f:=findf(p,q,r,k);
Error, (in findf) No polynomial f exists
> gosper(a,k);
Error, (in gosper) No hypergeometric term antidifference exists
> a:='a': s:='s': p:='p': q:='q': r:='r': f:='f':
>
```

Zeilberger's Algorithm

```
> with(sumtools);
Warning, these names have been redefined: Sumtohyper, extended_gosper,
gosper, hyperterm, simpcomb, sumrecursion
```

$$[Hypersum, Sumtohyper, extended\_gosper, gosper, hyperrecursion, hypersum, hyperterm, simpcomb, sumrecursion, sumtohyper]$$

```
> sumrecursion(k*binomial(n,k),k,s(n));
```

$$(n-1)\,\mathrm{s}(n) - 2\,n\,\mathrm{s}(n-1)$$

```
> sumrecursion((-1)^k*binomial(n,k)^2,k,s(n));
```

$$4\,(n-1)\,\mathrm{s}(n-2) + \mathrm{s}(n)\,n$$

```
> sumrecursion(binomial(n,k)^3,k,s(n));
```

$$-8(n-1)^2 s(n-2) - (7n^2-7n+2) s(n-1) + s(n) n^2$$

With Zeilberger's algorithm, we can do more complicated examples.

The Apéry numbers

```
> Sum(binomial(n,k)^2*binomial(n+k,k)^2,k=0..n);
```

$$\sum_{k=0}^{n} \mathrm{binomial}(n,k)^2 \,\mathrm{binomial}(n+k,k)^2$$

satisfy the recurrence equation

```
> sumrecursion(binomial(n,k)^2*binomial(n+k,k)^2,k,A(n));
```

$$(n-1)^3 A(n-2) - (2n-1)(17n^2-17n+5) A(n-1) + A(n) n^3$$

Four different representations of the Legendre polynomials:

(a) We consider the summand:

```
> legendre1:=binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k;
```

$$legendre1 := \mathrm{binomial}(n,k)\,\mathrm{binomial}(-n-1,k)\left(\frac{1}{2}-\frac{x}{2}\right)^k$$

The sum

```
> Sum(legendre1,k=0..n);
```

$$\sum_{k=0}^{n} \mathrm{binomial}(n,k)\,\mathrm{binomial}(-n-1,k)\left(\frac{1}{2}-\frac{x}{2}\right)^k$$

has the hypergeometric representation

```
> Sumtohyper(legendre1,k);
```

$$\mathrm{Hypergeom}\left([-n, 1+n], [1], \frac{1}{2}-\frac{x}{2}\right)$$

and satisfies the recurrence equation

```
> sumrecursion(legendre1,k,P(n));
```

$$(n-1) P(n-2) - (2n-1) x P(n-1) + P(n) n$$

(b) We consider the summand:

```
> legendre2:=1/2^n*binomial(n,k)^2*(x-1)^(n-k)*(x+1)^k;
```

$$legendre2 := \frac{\mathrm{binomial}(n,k)^2 (x-1)^{(n-k)} (x+1)^k}{2^n}$$

The sum

```
> Sum(legendre2,k=0..n);
```

$$\sum_{k=0}^{n} \frac{\mathrm{binomial}(n,k)^2 (x-1)^{(n-k)} (x+1)^k}{2^n}$$

has the hypergeometric representation

```
> Sumtohyper(legendre2,k);
```

$$\left(\frac{x}{2}-\frac{1}{2}\right)^n \mathrm{Hypergeom}\left([-n, -n], [1], \frac{x+1}{x-1}\right)$$

and satisfies the recurrence equation

```
> sumrecursion(legendre2,k,P(n));
```

$$(n-1)P(n-2)-(2n-1)xP(n-1)+P(n)n$$

(c) We consider the summand:

```
> legendre3:=1/2^n*(-1)^k*binomial(n,k)*binomial(2*n-2*k,n)*x^(
  n-2*k);
```

$$legendre3 := \frac{(-1)^k \operatorname{binomial}(n,k) \operatorname{binomial}(2n-2k,n)\, x^{(n-2k)}}{2^n}$$

The sum

```
> Sum(legendre3,k=0..floor(n/2));
```

$$\sum_{k=0}^{\operatorname{floor}\left(\frac{n}{2}\right)} \frac{(-1)^k \operatorname{binomial}(n,k) \operatorname{binomial}(2n-2k,n)\, x^{(n-2k)}}{2^n}$$

has the hypergeometric representation

```
> Sumtohyper(legendre3,k);
```

$$2^{(-n)} \operatorname{binomial}(2n,n)\, x^n \operatorname{Hypergeom}\left(\left[-\frac{n}{2}, -\frac{n}{2}+\frac{1}{2}\right], \left[-n+\frac{1}{2}\right], \frac{1}{x^2}\right)$$

and satisfies the recurrence equation

```
> sumrecursion(legendre3,k,P(n));
```

$$(n-1)P(n-2)-(2n-1)xP(n-1)+P(n)n$$

(d) We consider the summand:

```
> legendre4:=x^n*hyperterm([-n/2,(1-n)/2],[1],1-1/x^2,k);
```

$$legendre4 := \frac{x^n \operatorname{pochhammer}\left(-\frac{n}{2}, k\right) \operatorname{pochhammer}\left(-\frac{n}{2}+\frac{1}{2}, k\right) \left(1-\frac{1}{x^2}\right)^k}{(k!)^2}$$

The sum

```
> Sum(legendre4,k=0..floor(n/2));
```

$$\sum_{k=0}^{\operatorname{floor}\left(\frac{n}{2}\right)} \frac{x^n \operatorname{pochhammer}\left(-\frac{n}{2}, k\right) \operatorname{pochhammer}\left(-\frac{n}{2}+\frac{1}{2}, k\right) \left(1-\frac{1}{x^2}\right)^k}{(k!)^2}$$

has the hypergeometric representation

```
> Sumtohyper(legendre4,k);
```

$$x^n \operatorname{Hypergeom}\left(\left[-\frac{n}{2}, -\frac{n}{2}+\frac{1}{2}\right], [1], \frac{(x-1)(x+1)}{x^2}\right)$$

and satisfies the recurrence equation

```
> sumrecursion(legendre4,k,P(n));
```

$$(n-1)P(n-2)-(2n-1)xP(n-1)+P(n)n$$

Dougall's identity:

```
> TIME:=time():
  sumrecursion(hyperterm([a,1+a/2,b,c,d,1+2*a-b-c-d+n,-n],[a/2,
  1+a-b,1+a-c,1+a-d,b+c+d-a-n,1+a+n],1,k),k,s(n));
  time()-TIME;
```

$$-(a+n)(a-c-d+n)(a-b-d+n)(a-c+n-b)\,s(n-1) + s(n)(-d+a+n)(a-c+n)(a-b+n)(-b-c-d+a+n)$$

0.701

```
> term:=hyperterm([a,1+a/2,b,c,d,1+2*a-b-c-d+n,-n],[a/2,1+a-b,1+a-c,1+a-d,b+c+d-a-n,1+a+n],1,k);
```

$$term := \text{pochhammer}(a,k)\,\text{pochhammer}\left(1+\frac{a}{2},k\right)\text{pochhammer}(b,k)\,\text{pochhammer}(c,k)\,\text{pochhammer}(d,k)\,\text{pochhammer}(1+2a-b-c-d+n,k)\,\text{pochhammer}(-n,k) \Big/ \left(\text{pochhammer}\left(\frac{a}{2},k\right)\text{pochhammer}(1+a-b,k)\,\text{pochhammer}(1+a-c,k)\,\text{pochhammer}(1+a-d,k)\,\text{pochhammer}(b+c+d-a-n,k)\,\text{pochhammer}(1+a+n,k)\,k!\right)$$

```
> A:=term+sigma(1)*subs(n=n+1,term);
```

$$A := \text{pochhammer}(a,k)\,\text{pochhammer}\left(1+\frac{a}{2},k\right)\text{pochhammer}(b,k)\,\text{pochhammer}(c,k)\,\text{pochhammer}(d,k)\,\text{pochhammer}(1+2a-b-c-d+n,k)\,\text{pochhammer}(-n,k) \Big/ \left(\text{pochhammer}\left(\frac{a}{2},k\right)\text{pochhammer}(1+a-b,k)\,\text{pochhammer}(1+a-c,k)\,\text{pochhammer}(1+a-d,k)\,\text{pochhammer}(b+c+d-a-n,k)\,\text{pochhammer}(1+a+n,k)\,k!\right) + \sigma(1)\,\text{pochhammer}(a,k)\,\text{pochhammer}\left(1+\frac{a}{2},k\right)\text{pochhammer}(b,k)\,\text{pochhammer}(c,k)\,\text{pochhammer}(d,k)\,\text{pochhammer}(2+2a-b-c-d+n,k)\,\text{pochhammer}(-n-1,k) \Big/ \left(\text{pochhammer}\left(\frac{a}{2},k\right)\text{pochhammer}(1+a-b,k)\,\text{pochhammer}(1+a-c,k)\,\text{pochhammer}(1+a-d,k)\,\text{pochhammer}(b+c+d-a-1-n,k)\,\text{pochhammer}(2+a+n,k)\,k!\right)$$

```
> rat:=simpcomb(subs(k=k+1,A)/A);
```

$$rat := (n^4 - 8can + 4cbn - 8bna + 4cdn - 8adn + 4bdn - 5adn^2 + 2cdn^2 - 5bn^2a + 2cbn^2 - 5can^2 - 3na^2d + nc^2a + 2nabc + nb^2a - 3nca^2 + 2bdn^2 + nd^2a + 2ndca + 2ndab - 3na^2b - 3\sigma(1)nca^2 + \sigma(1)nb^2a + 2\sigma(1)bdn^2 - 3\sigma(1)na^2b + 7an - 4cn - 4bn - 4dn + 11an^2 + 7a^2n + 2b^2n + 2c^2n - 6cn^2 + 2d^2n - 6dn^2 - 6bn^2 - 2cn^3 + c^2n^2 + b^2n^2 + 5a^2n^2 + 4an^3 + 2na^3 - 2bn^3 - 2dn^3 + d^2n^2 - k^2 + 4n^3 - k^2n^2 - 2k^2n - 2kn^2 - 4nk + 5n^2 - 2bkda - 2bkca - 2ckda + 2n - 2k - \sigma(1)k^2n^2 + 2adnk + 2bnak + 2cank - d^2ka - b^2ka$$

$$- c^{2} k a - 3 a^{2} n k - a n^{2} k - 3 a k^{2} n - 2 b k^{2} c - 2 b k^{2} d + 2 b k^{2} n + 3 a^{2} k d + 3 a^{2} k c$$
$$+ 3 a k^{2} d + 3 a k^{2} c + 2 d k^{2} n + 2 c k^{2} n - 2 c k^{2} d + 3 a^{2} k b + 3 a k^{2} b + 4 d k n + 4 c k n$$
$$- 4 c k d + 8 a k b + 8 a k d + 8 a k c - 8 a k n - 4 b k d - 4 b k c + 4 b k n - 3 \sigma(1) a k$$
$$- 2 a^{3} k - 2 a^{2} k^{2} - b^{2} k^{2} - c^{2} k^{2} - d^{2} k^{2} - 3 a k^{2} + 2 b k^{2} + 2 c k^{2} + 2 d k^{2} - 7 a^{2} k - 2 b^{2} k$$
$$- 2 c^{2} k - 2 d^{2} k - 7 a k + 4 b k + 4 c k + 4 d k - \sigma(1) a^{2} n k + \sigma(1) n c^{2} a - 5 \sigma(1) b n^{2} a$$
$$+ 2 \sigma(1) c d n^{2} + 4 \sigma(1) b d n - 10 \sigma(1) a d n + 4 \sigma(1) c d n - 10 \sigma(1) b n a$$
$$+ 2 \sigma(1) d c a + 4 \sigma(1) c b n - 10 \sigma(1) c a n + 2 \sigma(1) d a b - 2 \sigma(1) k + 2 \sigma(1) n d a b$$
$$+ 2 \sigma(1) n a b c + 2 \sigma(1) n d c a - 2 \sigma(1) d + \sigma(1) n^{4} + 2 \sigma(1) a^{3} + \sigma(1) b^{2} + 4 \sigma(1) a^{2}$$
$$+ \sigma(1) c^{2} + \sigma(1) d^{2} - \sigma(1) k^{2} + 4 \sigma(1) n^{3} + 5 \sigma(1) n^{2} + 2 \sigma(1) a - 2 \sigma(1) b - 2 \sigma(1) c$$
$$+ 2 \sigma(1) n + \sigma(1) n d^{2} a - 5 \sigma(1) c a n^{2} + 2 \sigma(1) c b n^{2} - 3 \sigma(1) n a^{2} d - 5 \sigma(1) a d n^{2}$$
$$- 5 \sigma(1) a b + \sigma(1) d^{2} n^{2} - 2 \sigma(1) d n^{3} - 2 \sigma(1) b n^{3} - \sigma(1) a^{2} k - \sigma(1) a k^{2}$$
$$- 4 \sigma(1) n k - 2 \sigma(1) k n^{2} - 2 \sigma(1) k^{2} n + 4 \sigma(1) a n^{3} + 5 \sigma(1) a^{2} n^{2} + 2 \sigma(1) n a^{3}$$
$$- 2 \sigma(1) c n^{3} + \sigma(1) c^{2} n^{2} + \sigma(1) b^{2} n^{2} - 6 \sigma(1) b n^{2} - 6 \sigma(1) d n^{2} + \sigma(1) d^{2} a$$
$$+ 2 \sigma(1) d^{2} n - 3 \sigma(1) a^{2} d - 3 \sigma(1) a^{2} b + \sigma(1) b^{2} a + 2 \sigma(1) b^{2} n - 6 \sigma(1) c n^{2}$$
$$+ \sigma(1) c^{2} a + 2 \sigma(1) c^{2} n + 9 \sigma(1) a^{2} n + 11 \sigma(1) a n^{2} - 3 \sigma(1) c a^{2} - 6 \sigma(1) d n$$
$$+ 2 \sigma(1) c d - 6 \sigma(1) b n + 2 \sigma(1) b d - 5 \sigma(1) c a + 9 \sigma(1) a n - 6 \sigma(1) c n$$
$$- 5 \sigma(1) a d + 2 \sigma(1) c b + 2 \sigma(1) a b c - \sigma(1) a k^{2} n - 4 \sigma(1) a k n - \sigma(1) a n^{2} k)$$
$$(2 + a + 2 k) (a + k) (b + k) (c + k) (d + k) (1 + 2 a - b - c - d + n + k) (n - k + 1) \Big/ (($$
$$1 + 2 d a b + 2 c b - 5 c a + n^{4} - 10 c a n + 4 c b n + 2 d c a + 2 a^{3} - 10 b n a + 4 c d n$$
$$- 10 a d n + 4 b d n - 5 a d n^{2} + 2 c d n^{2} - 5 b n^{2} a + 2 c b n^{2} - 5 c a n^{2} - 3 n a^{2} d + n c^{2} a$$
$$+ 2 n a b c + n b^{2} a - 3 n c a^{2} + 2 b d n^{2} + n d^{2} a + 2 n d c a + 2 n d a b - 3 n a^{2} b + 2 b d$$
$$+ b^{2} - 3 \sigma(1) n c a^{2} + \sigma(1) n b^{2} a + 2 \sigma(1) b d n^{2} - 3 \sigma(1) n a^{2} b + 12 a n + 5 a^{2} - 5 a d$$
$$- 6 c n + 2 c d + c^{2} - 6 b n - 3 c a^{2} - 6 d n + d^{2} + 12 a n^{2} + 10 a^{2} n + 2 b^{2} n + b^{2} a + 2 c^{2} n$$
$$+ c^{2} a - 6 c n^{2} - 3 a^{2} d - 3 a^{2} b + 2 d^{2} n + d^{2} a - 6 d n^{2} - 6 b n^{2} - 2 c n^{3} + c^{2} n^{2} + b^{2} n^{2}$$
$$+ 5 a^{2} n^{2} + 4 a n^{3} + 2 n a^{3} - 2 b n^{3} - 2 d n^{3} + d^{2} n^{2} - k^{2} + 4 n^{3} - k^{2} n^{2} - 2 k^{2} n + 6 n^{2}$$
$$- 2 b k d a - 2 b k c a - 2 c k d a + 4 a - 2 b - 2 c + 4 n - 2 d - \sigma(1) k^{2} n^{2} + 2 a d n k$$
$$+ 2 b n a k + 2 c a n k + \sigma(1) - d^{2} k a - b^{2} k a - c^{2} k a - 3 a^{2} n k - a n^{2} k - 3 a k^{2} n$$
$$- 2 b k^{2} c - 2 b k^{2} d + 2 b k^{2} n + 3 a^{2} k d + 3 a^{2} k c + 3 a k^{2} d + 3 a k^{2} c + 2 d k^{2} n$$
$$+ 2 c k^{2} n - 2 c k^{2} d + 3 a^{2} k b + 3 a k^{2} b + 2 a k b + 2 a k d + 2 a k c - 2 a k n - \sigma(1) a k$$
$$- 2 a^{3} k - 2 a^{2} k^{2} - b^{2} k^{2} - c^{2} k^{2} - d^{2} k^{2} - 3 a k^{2} + 2 b k^{2} + 2 c k^{2} + 2 d k^{2} - 3 a^{2} k - a k$$
$$+ 2 a b c - 5 a b - \sigma(1) a^{2} n k + \sigma(1) n c^{2} a - 5 \sigma(1) b n^{2} a + 2 \sigma(1) c d n^{2}$$
$$+ 4 \sigma(1) b d n - 10 \sigma(1) a d n + 4 \sigma(1) c d n - 10 \sigma(1) b n a + 2 \sigma(1) d c a$$
$$+ 4 \sigma(1) c b n - 10 \sigma(1) c a n + 2 \sigma(1) d a b + 2 \sigma(1) n d a b + 2 \sigma(1) n a b c$$
$$+ 2 \sigma(1) n d c a - 2 \sigma(1) d + \sigma(1) n^{4} + 2 \sigma(1) a^{3} + \sigma(1) b^{2} + 5 \sigma(1) a^{2} + \sigma(1) c^{2}$$
$$+ \sigma(1) d^{2} - \sigma(1) k^{2} + 4 \sigma(1) n^{3} + 6 \sigma(1) n^{2} + 4 \sigma(1) a - 2 \sigma(1) b - 2 \sigma(1) c + 4 \sigma(1) n$$

$+\sigma(1)\, n\, d^2\, a - 5\,\sigma(1)\, c\, a\, n^2 + 2\,\sigma(1)\, c\, b\, n^2 - 3\,\sigma(1)\, n\, a^2\, d - 5\,\sigma(1)\, a\, d\, n^2$
$- 5\,\sigma(1)\, a\, b + \sigma(1)\, d^2\, n^2 - 2\,\sigma(1)\, d\, n^3 - 2\,\sigma(1)\, b\, n^3 - \sigma(1)\, a^2\, k - \sigma(1)\, a\, k^2$
$- 2\,\sigma(1)\, k^2\, n + 4\,\sigma(1)\, a\, n^3 + 5\,\sigma(1)\, a^2\, n^2 + 2\,\sigma(1)\, n\, a^3 - 2\,\sigma(1)\, c\, n^3 + \sigma(1)\, c^2\, n^2$
$+ \sigma(1)\, b^2\, n^2 - 6\,\sigma(1)\, b\, n^2 - 6\,\sigma(1)\, d\, n^2 + \sigma(1)\, d^2\, a + 2\,\sigma(1)\, d^2\, n - 3\,\sigma(1)\, a^2\, d$
$- 3\,\sigma(1)\, a^2\, b + \sigma(1)\, b^2\, a + 2\,\sigma(1)\, b^2\, n - 6\,\sigma(1)\, c\, n^2 + \sigma(1)\, c^2\, a + 2\,\sigma(1)\, c^2\, n$
$+ 10\,\sigma(1)\, a^2\, n + 12\,\sigma(1)\, a\, n^2 - 3\,\sigma(1)\, c\, a^2 - 6\,\sigma(1)\, d\, n + 2\,\sigma(1)\, c\, d - 6\,\sigma(1)\, b\, n$
$+ 2\,\sigma(1)\, b\, d - 5\,\sigma(1)\, c\, a + 12\,\sigma(1)\, a\, n - 6\,\sigma(1)\, c\, n - 5\,\sigma(1)\, a\, d + 2\,\sigma(1)\, c\, b$
$+ 2\,\sigma(1)\, a\, b\, c - \sigma(1)\, a\, k^2\, n - 2\,\sigma(1)\, a\, k\, n - \sigma(1)\, a\, n^2\, k)\,(k+1)$
$(-b - c - d + a + n - k)\,(1 + a - d + k)\,(1 + a - c + k)\,(1 + a - b + k)\,(a + 2\,k)$
$(2 + a + n + k))$

```
> closedform(term,k,n);
```

pochhammer($a+1, n$) pochhammer($1+a-c-d, n$) pochhammer($a-b-d+1, n$)
pochhammer($a-c+1-b, n$) / (pochhammer($1+a-d, n$) pochhammer($1+a-c, n$)
pochhammer($1+a-b, n$) pochhammer($-b-c-d+a+1, n$))

Proof of Clausen's formula by Cauchy product:

```
> summand:=j->hyperterm([a,b],[a+b+1/2],1,j);
```

$$summand := j \to \text{hyperterm}\left([a, b], \left[a + b + \frac{1}{2}\right], 1, j\right)$$

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

```
> Closedform(summand(j)*summand(k-j),j,k);
```

$$\text{Hyperterm}\left([2\,b, 2\,a, b + a], \left[2\,a + 2\,b, a + b + \frac{1}{2}\right], 1, k\right)$$

example from Joris van der Jeugt's talk: Whipple's transformation (2.10):

```
> RE1:=sumrecursion(subs(f=a+b+c-n+1-d-e,hyperterm([-n,a,b,c],[
  d,e,f],1,k)),k,S(n));
```

$RE1 := (1 + d + n)\,(n + 1 + e)\,(a + b + c - n - d - e)\,(a + b + c - n - 1 - d - e)\,\mathrm{S}(2 + n)$
$- (a + b + c - n - d - e)\,(-1 - d^2\, e - d\, e^2 + a - 3\,n - a\,b\,n - a\,c\,n - b\,c\,n + b - 2\,e - 2\,d$
$+ 2\,e\,a\,n + c - a\,b\,c - 2\,n\,e^2 - 2\,n\,d^2 + 2\,b\,n^2 + 2\,a\,n^2 - 4\,n^2\,e + 2\,c\,n^2 - 4\,d\,n^2 + 2\,a\,d\,n$
$- 2\,n^3 + 2\,b\,d\,n + d\,e\,b - 5\,d\,e\,n + 2\,c\,d\,n + 2\,e\,c\,n + 2\,e\,b\,n + d\,e\,c + d\,e\,a - a\,b - a\,c$
$- 4\,n^2 - 2\,e^2 + 2\,c\,d - 6\,n\,d - 6\,n\,e - 2\,d^2 - 5\,d\,e + 2\,c\,e + 2\,b\,e + 3\,c\,n - b\,c + 3\,b\,n$
$+ 2\,b\,d + 3\,a\,n + 2\,a\,e + 2\,a\,d)\,\mathrm{S}(1 + n)$
$+ (1 + n)\,(-b - c + n + d + e)\,(-n + a - e + c - d)\,(a - n + b - e - d)\,\mathrm{S}(n) = 0$

```
> RE2:=sumrecursion(subs(f=a+b+c-n+1-d-e,pochhammer(e-c,n)*poch
  hammer(f-c,n)/pochhammer(e,n)/pochhammer(f,n)*hyperterm([-n,d
  -a,d-b,c],[d,d+e-a-b,d+f-a-b],1,k)),k,S(n));
```

$RE2 := (1 + d + n)\,(n + 1 + e)\,(a + b + c - n - d - e)\,(a + b + c - n - 1 - d - e)\,\mathrm{S}(2 + n)$
$- (a + b + c - n - d - e)\,(-1 - d^2\, e - d\, e^2 + a - 3\,n - a\,b\,n - a\,c\,n - b\,c\,n + b - 2\,e - 2\,d$

$+2\,e\,a\,n+c-a\,b\,c-2\,n\,e^2-2\,n\,d^2+2\,b\,n^2+2\,a\,n^2-4\,n^2\,e+2\,c\,n^2-4\,d\,n^2+2\,a\,d\,n$
$-2\,n^3+2\,b\,d\,n+d\,e\,b-5\,d\,e\,n+2\,c\,d\,n+2\,e\,c\,n+2\,e\,b\,n+d\,e\,c+d\,e\,a-a\,b-a\,c$
$-4\,n^2-2\,e^2+2\,c\,d-6\,n\,d-6\,n\,e-2\,d^2-5\,d\,e+2\,c\,e+2\,b\,e+3\,c\,n-b\,c+3\,b\,n$
$+2\,b\,d+3\,a\,n+2\,a\,e+2\,a\,d)\,\mathrm{S}(1+n)$
$+(1+n)\,(-b-c+n+d+e)\,(-n+a-e+c-d)\,(a-n+b-e-d)\,\mathrm{S}(n)=0$

```
> op(1,RE1)-op(1,RE2);
```

$$0$$

```
>
```

A Generating Function Problem

```
> RE:=sumrecursion(binomial(alpha+n-1,n)*legendre4*z^n,n,s(k));
```

$$RE := -4\,(k+1)^2\,(x\,z-1)^2\,\mathrm{s}(k+1)+z^2\,(2\,k+\alpha+1)\,(2\,k+\alpha)\,(x-1)\,(x+1)\,\mathrm{s}(k)=0$$

```
> sol:=rsolve(RE,s(k));
```

$$sol := \frac{4^{(-k)}\,(z^2)^k\,(x-1)^k\,(x+1)^k\left(\frac{1}{(x\,z-1)^2}\right)^k\Gamma(2\,k+\alpha)\,\mathrm{s}(0)}{\Gamma(\alpha)\,\Gamma(k+1)^2}$$

We compute the initial value:

```
> s(0)=Sum(binomial(alpha+n-1,n)*subs(k=0,legendre4)*z^n,n=0..i
nfinity);
```

$$\mathrm{s}(0)=\sum_{n=0}^{\infty}\frac{\mathrm{binomial}(\alpha+n-1,n)\,x^n\,\mathrm{pochhammer}\left(-\frac{n}{2},0\right)\mathrm{pochhammer}\left(-\frac{n}{2}+\frac{1}{2},0\right)z^n}{(0!)^2}$$

```
> aw:=s(0)=sum(binomial(alpha+n-1,n)*subs(k=0,legendre4)*z^n,n=
0..infinity);
```

$$aw := \mathrm{s}(0)=\frac{1}{(1-x\,z)^\alpha}$$

Therefore we get the solution:

```
> sol:=subs(aw,sol);
```

$$sol := \frac{4^{(-k)}\,(z^2)^k\,(x-1)^k\,(x+1)^k\left(\frac{1}{(x\,z-1)^2}\right)^k\Gamma(2\,k+\alpha)}{\Gamma(\alpha)\,\Gamma(k+1)^2\,(1-x\,z)^\alpha}$$

which we put into hypergeometric form:

```
> Sumtohyper(sol,k);
```

$$(1-x\,z)^{(-\alpha)}\,\mathrm{Hypergeom}\left(\left[\frac{\alpha}{2},\frac{1}{2}+\frac{\alpha}{2}\right],[1],\frac{z^2\,(x-1)\,(x+1)}{(x\,z-1)^2}\right)$$

```
>
```

Infinite Sums

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

```
> read "infhsum.mpl";
Error, unable to read `infhsum.mpl`
```

Gauss identity

```
> infclosedform(hyperterm([a,b],[c],1,k),k,c);
```

$$\mathrm{infclosedform}\left(\frac{\mathrm{pochhammer}(a,k)\,\mathrm{pochhammer}(b,k)}{\mathrm{pochhammer}(c,k)\,k!}, k, c\right)$$

Kummer's identity

```
> infclosedform(hyperterm([a,b],[1+a-b],-1,k),k,a);
```

$$\mathrm{infclosedform}\left(\frac{\mathrm{pochhammer}(a,k)\,\mathrm{pochhammer}(b,k)\,(-1)^k}{\mathrm{pochhammer}(1+a-b,k)\,k!}, k, a\right)$$

Pfaff-Saalschütz identity

```
> infclosedform(hyperterm([a,b,c],[d,1+a+b+c-d],1,k),k,d);
```

$$\mathrm{infclosedform}\left(\frac{\mathrm{pochhammer}(a,k)\,\mathrm{pochhammer}(b,k)\,\mathrm{pochhammer}(c,k)}{\mathrm{pochhammer}(d,k)\,\mathrm{pochhammer}(1+a+b+c-d,k)\,k!}, k, d\right)$$

Note that this is an non-obvious generalization of the Pfaff-Saalschütz identity.

```
>
```

The WZ Method

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

```
> WZcertificate:=proc(F,n,k)
  local G;
  G:=gosper(subs(n=n+1,F)-F,k);
  simpcomb(G/F);
  end:
> WZcertificate(binomial(n,k),n,k);
Error, (in gosper) No hypergeometric term antidifference exists
> F:=binomial(n,k)/2^n;
```

$$F := \frac{\mathrm{binomial}(n,k)}{2^n}$$

```
> R:=WZcertificate(F,n,k);
```

$$R := -\frac{k}{2\,(n-k+1)}$$

Knowing this certificate function, we have only to check that two rational functions are equal:

```
> rationalproof:=(F,n,k,R)->[simpcomb(subs(n=n+1,F)/F)-1,subs(k
  =k+1,R)*simpcomb(subs(k=k+1,F)/F)-R]:
> rationalproof(F,n,k,R);
```

$$\left[\frac{1+n}{2\,(n-k+1)}-1,\,-\frac{1}{2}+\frac{k}{2\,(n-k+1)}\right]$$

Dougall's theorem:

```
> F:=hyperterm([a,1+a/2,b,c,d,1+2*a-b-c-d+n,-n],[a/2,1+a-b,1+a-c,1+a-d,b+c+d-a-n,1+a+n],1,k)/hyperterm([1+a,a+1-b-c,a+1-b-d,a+1-c-d,1],[1+a-b,1+a-c,1+a-d,1+a-b-c-d],1,n);
```

$$F := \mathrm{pochhammer}(a,k)\,\mathrm{pochhammer}\left(1+\frac{a}{2},k\right)\mathrm{pochhammer}(b,k)\,\mathrm{pochhammer}(c,k)\,\mathrm{pochhammer}(d,k)\,\mathrm{pochhammer}(1+2\,a-b-c-d+n,k)\,\mathrm{pochhammer}(-n,k)\,\mathrm{pochhammer}(1+a-b,n)\,\mathrm{pochhammer}(1+a-c,n)\,\mathrm{pochhammer}(1+a-d,n)\,\mathrm{pochhammer}(1+a-b-c-d,n)\Big/\Big(\mathrm{pochhammer}\left(\frac{a}{2},k\right)\mathrm{pochhammer}(1+a-b,k)\,\mathrm{pochhammer}(1+a-c,k)\,\mathrm{pochhammer}(1+a-d,k)\,\mathrm{pochhammer}(b+c+d-a-n,k)\,\mathrm{pochhammer}(1+a+n,k)\,k!\,\mathrm{pochhammer}(1+a,n)\,\mathrm{pochhammer}(a+1-b-c,n)\,\mathrm{pochhammer}(a+1-b-d,n)\,\mathrm{pochhammer}(a+1-c-d,n)\Big)$$

```
> R:=WZcertificate(F,n,k);
```

$$R := -(-b-c-d+a+n-k+1)\,k\,(a-d+k)\,(a-c+k)\,(a-b+k)\,(2\,n+2+2\,a-b-c-d)/((n-k+1)\,(1+n+a-c-d)\,(a-b+n-d+1)\,(a-c+1+n-b)\,(1+2\,a-b-c-d+n)\,(a+2\,k))$$

```
> proof:=rationalproof(F,n,k,R);
```

$$\mathit{proof} := [(1+2\,a-b-c-d+n+k)\,(a-b+n+1)\,(a+n-c+1)\,(n-d+a+1)\,(1+n)\,(-b-c-d+a+n-k+1)/((n-k+1)\,(1+2\,a-b-c-d+n)\,(1+a+n+k)\,(a-c+1+n-b)\,(a-b+n-d+1)\,(1+n+a-c-d))-1,\,-(2\,n+2+2\,a-b-c-d)\,(a+k)\,(b+k)\,(c+k)\,(d+k)\,(1+2\,a-b-c-d+n+k)/((1+n+a-c-d)\,(a-b+n-d+1)\,(a-c+1+n-b)\,(1+2\,a-b-c-d+n)\,(a+2\,k)\,(1+a+n+k))+(-b-c-d+a+n-k+1)\,k\,(a-d+k)\,(a-c+k)\,(a-b+k)\,(2\,n+2+2\,a-b-c-d)/((n-k+1)\,(1+n+a-c-d)\,(a-b+n-d+1)\,(a-c+1+n-b)\,(1+2\,a-b-c-d+n)\,(a+2\,k))]$$

```
> normal(op(1,proof)-op(2,proof));
```

$$0$$

```
>
```

Differential Equations for Hypergeometric Sums

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

The differential equation of the sine function:

```
> sumdiffeq((-1)^k/(2*k+1)!*x^(2*k+1),k,s(x));
```

$$s(x)+\left(\frac{d^2}{dx^2}s(x)\right)=0$$

The four different hypergeometric representations of the Legendre polynomials all lead to the same differential equation:

```
> legendre1:=binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k;
  sumdiffeq(legendre1,k,P(x));
```

$$legendre1 := \text{binomial}(n,k)\,\text{binomial}(-n-1,k)\left(\frac{1}{2}-\frac{x}{2}\right)^k$$

$$-(x-1)(x+1)\left(\frac{d^2}{dx^2}P(x)\right)-2x\left(\frac{d}{dx}P(x)\right)+P(x)\,n\,(1+n)=0$$

```
> legendre2:=1/2^n*binomial(n,k)^2*(x-1)^(n-k)*(x+1)^k;
  sumdiffeq(legendre2,k,P(x));
```

$$legendre2 := \frac{\text{binomial}(n,k)^2\,(x-1)^{(n-k)}\,(x+1)^k}{2^n}$$

$$-(x-1)(x+1)\left(\frac{d^2}{dx^2}P(x)\right)-2x\left(\frac{d}{dx}P(x)\right)+P(x)\,n\,(1+n)=0$$

```
> legendre3:=1/2^n*(-1)^k*binomial(n,k)*binomial(2*n-2*k,n)*x^(
  n-2*k); sumdiffeq(legendre3,k,P(x));
```

$$legendre3 := \frac{(-1)^k\,\text{binomial}(n,k)\,\text{binomial}(2n-2k,n)\,x^{(n-2k)}}{2^n}$$

$$-(x-1)(x+1)\left(\frac{d^2}{dx^2}P(x)\right)-2x\left(\frac{d}{dx}P(x)\right)+P(x)\,n\,(1+n)=0$$

```
> legendre4:=x^n*hyperterm([-n/2,(1-n)/2],[1],1-1/x^2,k);
  sumdiffeq(legendre4,k,P(x));
```

$$legendre4 := \frac{x^n\,\text{pochhammer}\left(-\frac{n}{2},k\right)\text{pochhammer}\left(-\frac{n}{2}+\frac{1}{2},k\right)\left(1-\frac{1}{x^2}\right)^k}{(k!)^2}$$

$$-(x-1)(x+1)\left(\frac{d^2}{dx^2}P(x)\right)-2x\left(\frac{d}{dx}P(x)\right)+P(x)\,n\,(1+n)=0$$

```
>
```

Petkovsek's Algorithm

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

For the following sum Zeilberger's algorithm finds a recurrence equation of order *c*-1 instead of 1:

```
> Sum((-1)^k*binomial(n,k)*binomial(c*k,n),k=0..n)=(-c)^k;
```

$$\sum_{k=0}^{n} (-1)^k \operatorname{binomial}(n,k) \operatorname{binomial}(c\,k,n) = (-c)^k$$

We compute:

```
> rec:=sumrecursion((-1)^k*binomial(n,k)*binomial(4*k,n),k,s(n)
  );
```

$$rec := 3\,(3\,n+7)\,(3\,n+4)\,(3\,n+8)\,s(n+3) + 4\,(3\,n+4)\,(37\,n^2+180\,n+218)\,s(2+n) + 16\,(2+n)\,(33\,n^2+125\,n+107)\,s(1+n) + 64\,(3\,n+7)\,(2+n)\,(1+n)\,s(n) = 0$$

```
> TIME:=time():
  rechyper(rec,s(n));
  time()-TIME;
```

$$\{-4\}$$

$$1.572$$

We load a package which includes an implementation of a faster algorithm by Mark van Hoeij:

```
> libname:=`C:/Dokumente und Einstellungen/Koepf/Eigene
  Dateien/Koepf/Vorträge/SummerSchool`,libname;
```

libname := "C:/Dokumente und Einstellungen/Koepf/Eigene Dateien/Koepf/Vorträge/Sum\
merSchool", "C:\Programme\Maple 8/lib", "C:/Dokumente und Einstellungen/koepf/Eigen\
e Dateien/Koepf/Vorträge/SummerSchool/hsum"

```
> TIME:=time():
  `LREtools/hsols`(rec,s(n));
  time()-TIME;
```

$$[(-4)^n]$$

$$0.461$$

For c=5, we get

```
> rec:=sumrecursion((-1)^k*binomial(n,k)*binomial(5*k,n),k,s(n)
  );
```

$$rec := 8\,(2\,n+7)\,(5+2\,n)\,(13+4\,n)\,(4\,n+9)\,(4\,n+5)\,(15+4\,n)\,s(4+n) + 5\,(9\,n+31)\,(4\,n+9)\,(4\,n+5)\,(5+2\,n)\,(41\,n^2+283\,n+486)\,s(n+3) + 25\,(4\,n+5)\,(n+3)\,(1048\,n^4+12242\,n^3+52919\,n^2+100279\,n+70302)\,s(2+n) + 125\,(13+4\,n)\,(n+3)\,(2+n)\,(152\,n^3+1098\,n^2+2437\,n+1623)\,s(1+n) + 625\,(2\,n+7)\,(13+4\,n)\,(4\,n+9)\,(n+3)\,(2+n)\,(1+n)\,s(n) = 0$$

The following computation has to be interrupted:

```
> TIME:=time():
  rechyper(rec,s(n));
  time()-TIME;
Warning, computation interrupted
```

```
> TIME:=time():
  `LREtools/hsols`(rec,s(n));
  time()-TIME;
```

$$[(-5)^n]$$

$$0.370$$

Wolfram Koepf: Hypergeometric Summation, Exercise 9.3 (a):

```
> rec:=
sumrecursion(hyperterm([-n,a,a+1/2,b],[2*a,(b-n+1)/2,(b-n)/2+
1],1,k),k,s(n));
```

$$rec := (n+b+2)(-b+n+1)(-b+n)(n+2a+1)\,s(2+n)$$
$$-2(n+b+1)(-b+n)(n+a+1)(1+2a+n-b)\,s(1+n)$$
$$+(1+n)(n+b)(1+2a+n-b)(2a+n-b)\,s(n)=0$$

```
> TIME:=time():
res1:=rechyper(rec,s(n));
time()-TIME;
```

$$res1 := \{-\frac{(b+n)(-b+n+2a)}{(b+1+n)(b-n)}, -\frac{(1+n)(b+n)(-b+n+2a)}{(b+1+n)(b-n)(n+2a)}\}$$

$$4.336$$

```
> TIME:=time():
res2:=`LREtools/hsols`(rec,s(n));
time()-TIME;
```

$$res2 := \left[\frac{\Gamma(1+n)\,\Gamma(2a+n-b)}{\Gamma(n+2a)\,\Gamma(-b+n)(n+b)}, \frac{\Gamma(2a+n-b)}{\Gamma(-b+n)(n+b)}\right]$$

$$0.450$$

```
>
```

Hyperexponential Integration

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

Continuous version of Gosper's algorithm.

Does the function

```
> f:=exp(x^2);
```

$$f := e^{(x^2)}$$

have a hyperexponential antiderivative? The answer is

```
> contgosper(exp(x^2),x);
Error, (in contgosper) No hyperexponential antiderivative exists
```

The situation is different for

```
> contgosper(x*exp(x^2),x);
```

$$\frac{1}{2}e^{(x^2)}$$

Let's do a more complicated example:

```
> term:=diff((1+x^2)/(1-x^10),x);
```

$$term := \frac{2x}{1-x^{10}} + \frac{10(x^2+1)x^9}{(1-x^{10})^2}$$

```
> res:=contgosper(term,x);
```

$$res := -\frac{x^2+1}{(x^6-x^5+x-1)(x^4+x^3+x^2+x+1)}$$

```
> res:=normal(res);
```

$$res := -\frac{x^2+1}{(x^6-x^5+x-1)(x^4+x^3+x^2+x+1)}$$

```
> res:=normal(res,expanded);
```

$$res := \frac{-x^2-1}{-1+x^{10}}$$

Let's check Maple's internal integrator:

```
> res:=int(term,x);
```

$$res := -\frac{2}{5}\frac{\arctan\left(\frac{-4x+1+\sqrt{5}}{\sqrt{10-2\sqrt{5}}}\right)}{\sqrt{10-2\sqrt{5}}} - \frac{4}{5}\frac{\arctan\left(\frac{-4x+1+\sqrt{5}}{\sqrt{10-2\sqrt{5}}}\right)\sqrt{5}}{(10-2\sqrt{5})^{(3/2)}}$$
$$+ \frac{4\arctan\left(\frac{-4x+1+\sqrt{5}}{\sqrt{10-2\sqrt{5}}}\right)}{(10-2\sqrt{5})^{(3/2)}} - \frac{4\arctan\left(\frac{4x-1+\sqrt{5}}{\sqrt{10+2\sqrt{5}}}\right)}{(10+2\sqrt{5})^{(3/2)}} + \frac{2}{5}\frac{\arctan\left(\frac{4x-1+\sqrt{5}}{\sqrt{10+2\sqrt{5}}}\right)}{\sqrt{10+2\sqrt{5}}}$$
$$- \frac{4}{5}\frac{\arctan\left(\frac{4x-1+\sqrt{5}}{\sqrt{10+2\sqrt{5}}}\right)\sqrt{5}}{(10+2\sqrt{5})^{(3/2)}} - \frac{1}{5(x-1)}$$
$$+ \frac{(-8\sqrt{5}-(\sqrt{5}-5)(\sqrt{5}-1))x+2\sqrt{5}(\sqrt{5}-1)-20+4\sqrt{5}}{5(10+2\sqrt{5})(-2x^2-x+\sqrt{5}x-2)}$$
$$+ \frac{(-8\sqrt{5}-(-\sqrt{5}-5)(\sqrt{5}+1))x-2\sqrt{5}(\sqrt{5}+1)+4\sqrt{5}+20}{5(10-2\sqrt{5})(2x^2+x+\sqrt{5}x+2)}$$
$$+ \frac{2}{5}\frac{\arctan\left(\frac{-4x-1+\sqrt{5}}{\sqrt{10+2\sqrt{5}}}\right)}{\sqrt{10+2\sqrt{5}}} - \frac{2}{5}\frac{\arctan\left(\frac{4x+1+\sqrt{5}}{\sqrt{10-2\sqrt{5}}}\right)}{\sqrt{10-2\sqrt{5}}} + \frac{1}{5(x+1)}$$
$$+ \frac{(-8\sqrt{5}-(-\sqrt{5}-5)(\sqrt{5}+1))x+2\sqrt{5}(\sqrt{5}+1)-20-4\sqrt{5}}{5(10-2\sqrt{5})(-2x^2+x+\sqrt{5}x-2)}$$
$$+ \frac{(-8\sqrt{5}-(\sqrt{5}-5)(\sqrt{5}-1))x-2\sqrt{5}(\sqrt{5}-1)-4\sqrt{5}+20}{5(10+2\sqrt{5})(2x^2-x+\sqrt{5}x+2)}$$
$$- \frac{4}{5}\frac{\arctan\left(\frac{-4x-1+\sqrt{5}}{\sqrt{10+2\sqrt{5}}}\right)\sqrt{5}}{(10+2\sqrt{5})^{(3/2)}} - \frac{4\arctan\left(\frac{-4x-1+\sqrt{5}}{\sqrt{10+2\sqrt{5}}}\right)}{(10+2\sqrt{5})^{(3/2)}}$$

$$-\frac{4}{5}\,\frac{\arctan\left(\frac{4x+1+\sqrt{5}}{\sqrt{10-2\sqrt{5}}}\right)\sqrt{5}}{(10-2\sqrt{5})^{(3/2)}}+\frac{4\arctan\left(\frac{4x+1+\sqrt{5}}{\sqrt{10-2\sqrt{5}}}\right)}{(10-2\sqrt{5})^{(3/2)}}$$

```
> res:=normal(res);
```

$$res := 320\,(x^2+1)\Big/\big((x-1)(5+\sqrt{5})(-2x^2-x+\sqrt{5}\,x-2)(\sqrt{5}-5)\\ (2x^2+x+\sqrt{5}\,x+2)(x+1)(-2x^2+x+\sqrt{5}\,x-2)(2x^2-x+\sqrt{5}\,x+2)\big)$$

```
> res:=normal(res,expanded);
```

$$res := \frac{-x^2-1}{-1+x^{10}}$$

Let's check Risch's algorithm:

```
> `int/risch`(term,x);
```

$$\frac{2\left(-\frac{1}{2}-\frac{x^2}{2}\right)}{-1+x^{10}}$$

```
>
```

Differential and Recurrence Equations for Integrals

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

A Beta type integral. We would like to compute:

```
> Int(t^x*(1-t)^y,t=0..1);
```

$$\int_0^1 t^x\,(1-t)^y\,dt$$

```
> integrand:=t^x*(1-t)^y;
```

$$integrand := t^x\,(1-t)^y$$

The integrand is a hyperexponential term:

```
> contratio(integrand,t);
```

$$\frac{-x+x\,t+y\,t}{t\,(-1+t)}$$

What type of result should we expect?

```
> [ratio(integrand,x),contratio(integrand,x)];
```

$$[t, \ln(t)]$$

```
> [ratio(integrand,y),contratio(integrand,y)];
```

$$[1-t, \ln(1-t)]$$

Application of the continuous version of Zeilberger's algorithm:

```
> REx:=intrecursion(integrand,t,S(x));
```

$$REx := -(x+y+2)\,\mathrm{S}(x+1)+(x+1)\,\mathrm{S}(x)=0$$

```
> rsolve(REx,S(x));
```

$$\frac{\Gamma(y+2)\,\Gamma(x+1)\,S(0)}{\Gamma(x+y+2)}$$

```
> REy:=intrecursion(integrand,t,S(y));
```

$$REy := -(x+y+2)\,S(y+1)+(y+1)\,S(y)=0$$

```
> rsolve(REy,S(y));
```

$$\frac{\Gamma(2+x)\,\Gamma(y+1)\,S(0)}{\Gamma(x+y+2)}$$

Therefore the integral should be a multiple of

```
> res:=GAMMA(x+1)*GAMMA(y+1)/GAMMA(x+y+2);
```

$$res := \frac{\Gamma(x+1)\,\Gamma(y+1)}{\Gamma(x+y+2)}$$

We let Maple do the integration:

```
> int(integrand,t=0..1);
Definite integration: Can't determine if the integral is convergent.
Need to know the sign of --> -y
Will now try indefinite integration and then take limits.
```

$$\frac{\Gamma(x+1)\,\Gamma(y+1)}{\Gamma(x+y+2)}$$

Another example. We would like to compute:

```
> Int(x^2/(x^4+t^2)/(1+t^2),t=0..infinity);
```

$$\int_0^{\infty} \frac{x^2}{(x^4+t^2)\,(1+t^2)}\,dt$$

```
> integrand:=x^2/(x^4+t^2)/(1+t^2);
```

$$integrand := \frac{x^2}{(x^4+t^2)\,(1+t^2)}$$

The integrand is a hyperexponential term:

```
> contratio(integrand,t);
```

$$-\frac{2\,t\,(1+2\,t^2+x^4)}{(1+t^2)\,(x^4+t^2)}$$

What type of result should we expect?

```
> [ratio(integrand,x),contratio(integrand,x)];
```

$$\left[\frac{(x+1)^2\,(x^4+t^2)}{(x^4+4\,x^3+6\,x^2+4\,x+1+t^2)\,x^2}, \frac{2\,(t-x^2)\,(x^2+t)}{x\,(x^4+t^2)}\right]$$

Application of the continuous version of Zeilberger's algorithm:

```
> RE:=intrecursion(integrand,t,S(x));
Error, (in intrecursion) Algorithm finds no recurrence equation of order <=
5
```

```
> DE:=intdiffeq(integrand,t,S(x));
```

$$DE := (x-1)\,(x+1)\,(x^2+1)\left(\frac{d^2}{dx^2}S(x)\right)x+(1+7\,x^4)\left(\frac{d}{dx}S(x)\right)+8\,S(x)\,x^3=0$$

```
> dsolve(DE,S(x));
```

$$S(x) = \frac{\_C1}{(x-1)(x+1)(x^2+1)} + \frac{\_C2\, x^2}{(x-1)(x+1)(x^2+1)}$$

```
> res:=int(integrand,t=0..infinity);
```

$$res := -\frac{1}{2}\,\frac{\pi\,(-x^2 + \mathrm{csgn}(\overline{x^2}))}{-1 + x^4}$$

```
> assume(x>0):
> res:=normal(res);
```

$$res := \frac{\pi}{2\,(x\sim^2 + 1)}$$

Which recurrence equation is valid for the result S(x)?

```
> ratio(res,x);
```

$$\frac{x\sim^2 + 1}{x\sim^2 + 2\,x\sim + 2}$$

```
> rat:=factor(ratio(res,x),I);
```

$$rat := \frac{(x\sim - I)\,(x\sim + I)}{(x\sim + 1 + I)\,(x\sim + 1 - I)}$$

Hence the recurrence equation for S(x) is

```
> denom(rat)*S(x+1)-numer(rat)*S(x)=0;
```

$$(x\sim + 1 + I)\,(x\sim + 1 - I)\,\mathrm{S}(x\sim + 1) - (x\sim - I)\,(x\sim + I)\,\mathrm{S}(x\sim) = 0$$

```
> x:='x':
```

An example from Olde Daalhuis' talk:

```
> intdiffeq(exp(-lambda*t)*t^alpha/(z-t),t,F(z));
```

$$z\left(\frac{d^2}{dz^2}\mathrm{F}(z)\right) + (\lambda\, z + 1 - \alpha)\left(\frac{d}{dz}\mathrm{F}(z)\right) + \mathrm{F}(z)\,\lambda = 0$$

```
> intdiffeq(exp(-lambda*t)*t^alpha/(z-t),t,F(lambda));
```

$$\lambda\left(\frac{d^2}{d\lambda^2}\mathrm{F}(\lambda)\right) + (1 + \lambda\, z + \alpha)\left(\frac{d}{d\lambda}\mathrm{F}(\lambda)\right) + \mathrm{F}(\lambda)\, z\,(1 + \alpha) = 0$$

```
>
```

Rodrigues Formulas

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

Rodrigues formula of the Legendre polynomials

```
> P(n,x)=(-1)^n/2^n/n!*diff((1-x^2)^n,x$n);
```

$$\mathrm{P}(n, x) = \frac{(-1)^n \left(\frac{\partial^n}{\partial x^n}(1 - x^2)^n\right)}{2^n\, n!}$$

The following function computes the recurrence equation of the family by Cauchy's integral formula

```
> RE:=rodriguesrecursion((-1)^n/2^n/n!,(1-x^2)^n,x,P(n));
```

$$RE := (2+n)\,P(2+n) - x\,(3+2\,n)\,P(1+n) + (1+n)\,P(n) = 0$$

Similarly, we get the differential equation

```
> DE:=rodriguesdiffeq((-1)^n/2^n/n!,(1-x^2)^n,n,P(x));
```

$$DE := -(x-1)\,(x+1)\left(\frac{d^2}{dx^2}P(x)\right) - 2\,x\left(\frac{d}{dx}P(x)\right) + P(x)\,n\,(1+n) = 0$$

The holonomic recurrence equation defines the Legendre polynomials uniquely up to the initial values

```
> P(0,x)=simplify(subs(n=0,(-1)^n/2^n/n!*(1-x^2)^n));
```

$$P(0, x) = 1$$

and

```
> P(1,x)=simplify(subs(n=1,(-1)^n/2^n/n!*diff((1-x^2)^n,x$n)));
```

$$P(1, x) = x$$

Rodrigues formula of the generalized Laguerre polynomials

```
> L(n,alpha,x)=exp(x)/n!/x^alpha*diff(exp(-x)*x^(alpha+n),x$n);
```

$$L(n, \alpha, x) = \frac{\mathbf{e}^{x}\left(\frac{\partial^n}{\partial x^n}\left(\mathbf{e}^{(-x)}\,x^{(n+\alpha)}\right)\right)}{n!\,x^{\alpha}}$$

The following function computes the recurrence equation of the family by Cauchy's integral formula

```
> RE:=rodriguesrecursion(exp(x)/n!/x^alpha,exp(-x)*x^(alpha+n),
  x,L(n));
```

$$RE := (2+n)\,L(2+n) - (3-x+2\,n+\alpha)\,L(1+n) + (1+n+\alpha)\,L(n) = 0$$

Similarly, we get the differential equation

```
> DE:=rodriguesdiffeq(exp(x)/n!/x^alpha,exp(-x)*x^(alpha+n),n,L
  (x));
```

$$DE := x\left(\frac{d^2}{dx^2}L(x)\right) - (x-\alpha-1)\left(\frac{d}{dx}L(x)\right) + L(x)\,n = 0$$

The holonomic recurrence equation defines the Legendre polynomials uniquely up to the initial values

```
> L(0,alpha,x)=simplify(subs(n=0,exp(x)/n!/x^alpha*exp(-x)*x^(a
  lpha+n)));
```

$$L(0, \alpha, x) = 1$$

and

```
> L(1,alpha,x)=simplify(subs(n=1,exp(x)/n!/x^alpha*diff((exp(-x
  )*x^(alpha+n),x$n))));
```

$$L(1, \alpha, x) = -x + \alpha + 1$$

An exapmple from Margit Rösler's talk: Hermite polynomials:

```
> rodriguesdiffeq((-1)^n*exp(x^2),exp(-x^2),n,H(x));
```

$$2\,H(x)\,n - 2\,x\left(\frac{d}{dx}H(x)\right) + \left(\frac{d^2}{dx^2}H(x)\right) = 0$$

```
> rodriguesrecursion((-1)^n*exp(x^2),exp(-x^2),x,H(n));
```

$$H(2+n) - 2\,x\,H(1+n) + 2\,H(n)\,(1+n) = 0$$

```
>
```

Generating Functions

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

The generating function of the generalized Laguerre polynomials satisfies the recurrence equation

```
> GFrecursion((1-z)^(-alpha-1)*exp((x*z)/(z-1)),1,z,L(n));
```

$$(2+n)\,L(2+n) - (3 - x + 2\,n + \alpha)\,L(1+n) + (1+n+\alpha)\,L(n) = 0$$

compare:

```
> RE;
```

$$(2+n)\,L(2+n) - (3 - x + 2\,n + \alpha)\,L(1+n) + (1+n+\alpha)\,L(n) = 0$$

and the differential equation

```
> GFdiffeq((1-z)^(-alpha-1)*exp((x*z)/(z-1)),1,z,n,L(x));
```

$$x\left(\frac{d^2}{dx^2}L(x)\right) - (x-\alpha-1)\left(\frac{d}{dx}L(x)\right) + L(x)\,n = 0$$

compare:

```
> DE;
```

$$x\left(\frac{d^2}{dx^2}L(x)\right) - (x-\alpha-1)\left(\frac{d}{dx}L(x)\right) + L(x)\,n = 0$$

The initial values:

```
> series((1-z)^(-alpha-1)*exp((x*z)/(z-1)),z=0,3);
```

$$1 + (\alpha - x + 1)\,z + \left(-x + \frac{x^2}{2} - \frac{(1+\alpha)\,(-\alpha-2)}{2} - (1+\alpha)\,x\right)z^2 + O(z^3)$$

```
>
```

q-Hypergeometric Series

```
> read "qsum6.mpl";
```

Package "q-Hypergeometric Summation", Maple 8

Copyright 1998-2002, Harald Böing & Wolfram Koepf, University of Kassel

special q-expressions

```
> qsimpcomb(qpochhammer(k,q,infinity));
```

$$\text{qpochhammer}(k, q, \infty)$$

```
> qsimpcomb(qfactorial(k,q));
```

$$\frac{\text{qpochhammer}(q, q, k)}{(1-q)^k}$$

```
> qsimpcomb(qGAMMA(z,q));
```

$$-\frac{\mathrm{qpochhammer}(q, q, \infty)\,(-1+q)}{\mathrm{qpochhammer}(q^z, q, \infty)\,(1-q)^z}$$

```
> qsimpcomb(qbinomial(n,k,q));
```

$$\frac{\mathrm{qpochhammer}(q, q, n)}{\mathrm{qpochhammer}(q, q, k)\,\mathrm{qpochhammer}(q, q, n-k)}$$

```
> qsimpcomb(qbrackets(k,q));
```

$$\frac{q^k-1}{-1+q}$$

q-Chu-Vandermonde Theorem

```
> RE:=qsumrecursion(qhyperterm([q^(-n),b],[c],q,c*q^n/b,k),q,k,
  S(n));
```

$$RE := b\,(c\,q^n - q)\,\mathrm{S}(n) + (-c\,q^n + b\,q)\,\mathrm{S}(n-1) = 0$$

```
> qsumrecursion(qhyperterm([q^(-n),b],[c],q,c*q^n/b,k),q,k,S(n)
  ,rec2qhyper=true);
```

$$\left[\mathrm{S}(n) = \frac{\mathrm{qpochhammer}\left(\frac{c}{b}, q, n\right)}{\mathrm{qpochhammer}(c, q, n)},\ 0 \le n\right]$$

q-orthogonal polynomials: Little q-Legendre polynomials

```
> qsumrecursion(qhyperterm([q^(-n),q^(n+1)],[q],q,q*x,k),q,k,P(
  n));
```

$$q^n\,(-1+q^n)\,(q+q^n)\,\mathrm{P}(n) - (q-q^{(2\,n)})\,(x\,q^{(1+n)} + q\,x - 2\,q^n + x\,q^n + x\,q^{(2\,n)})\,\mathrm{P}(n-1)$$
$$-\,(q^n+1)\,(q-q^n)\,q^n\,\mathrm{P}(n-2) = 0$$

```
> qsumdiffeq(qhyperterm([q^(-n),q^(n+1)],[q],q,q*x,k),q,k,P(x))
  ;
```

$$q\,(-1+q^n)\,(q^n\,q-1)\,\mathrm{P}(x)$$
$$-\,(-1+q)\,(q^3\,x\,q^n - q^2\,x\,(q^n)^2 + q^2\,x\,q^n - q\,x - q^n\,q + q^n)\,Dq_x(\mathrm{P}(x))$$
$$-\,x\,q^n\,(-1+q)^2\,(q^2\,x-1)\,q\,Dq_{x,\,x}(\mathrm{P}(x)) = 0$$

Big q-Legendre polynomials

```
> qsumrecursion(qhyperterm([q^(-n),q^(n+1),x],[q,c*q],q,q,k),q,
  k,P(n));
```

$$q\,(-1+q^n)\,(c\,q^n-1)\,(q+q^n)\,\mathrm{P}(n)$$
$$-\,(x\,q^{(1+n)} + q\,x - 2\,c\,q^{(1+n)} - 2\,q^{(1+n)} + x\,q^n + x\,q^{(2\,n)})\,(q-q^{(2\,n)})\,\mathrm{P}(n-1)$$
$$-\,q^n\,(q^n+1)\,(q-q^n)\,(c\,q-q^n)\,\mathrm{P}(n-2) = 0$$

```
> qsumdiffeq(qhyperterm([q^(-n),q^(n+1),x],[q,c*q],q,q,k),q,k,P
  (x));
```

$$(-1+q^n)\,(q^n\,q-1)\,\mathrm{P}(x)$$
$$-\,(-1+q)\,(q\,x\,q^n + q^2\,x\,q^n + c\,q^n - q\,q^n\,c - x - q^n\,q + q^n - q\,x\,(q^n)^2)\,Dq_x(\mathrm{P}(x))$$
$$-\,(-1+q)^2\,(q\,x-1)\,(q\,x-c)\,q^n\,Dq_{x,\,x}(\mathrm{P}(x)) = 0$$

or in other form:

```
> qsumdiffeq(qhyperterm([q^(-n),q^(n+1),x],[q,c*q],q,q,k),q,k,P(x),evalqdiff=true);
```

$$-(x-1)(x-c)q^n q\,P(x) + (q^2 x^2 (q^n)^2 - 2qxq^n + qq^n c - 2qxq^n c + qx^2 + cq^n)\,P(qx) - q^n(qx-1)(qx-c)\,P(q^2 x) = 0$$

examples from Dennis Stanton's lecture:

```
> qsumrecursion(qbinomial(N+k-1,k,q)*x^k*q^k,q,k,S(N),rec2qhyper=true);
```

$$\left[S(N) = \frac{S(0)}{\text{qpochhammer}(q\,x, q, N)}, 0 \le N\right]$$

```
> qsumrecursion(qbinomial(N,m,q)*x^m*y^(N-m),q,m,S(N),rec2qhyper=true);
```

$$[S(N) = \text{qpochhammer}(q, q, N)\,(-x\,y)^N, 0 \le N]$$

```
>
```

Orthogonal Polynomial Solutions of Recurrence Equations

```
> read "hsum6.mpl";
```

Package "Hypergeometric Summation", Maple 6

Copyright 2001, Wolfram Koepf, University of Kassel

```
> read "retode.mpl";
```

Package "REtoDE", Maple V - Maple 8

Copyright 2000-2002, Wolfram Koepf, University of Kassel

First example

```
> RE:=P(n+2)-(x-n-1)*P(n+1)+alpha*(n+1)^2*P(n)=0;
```

$$RE := P(2+n) - (x-n-1)\,P(1+n) + \alpha\,(1+n)^2\,P(n) = 0$$

```
> REtoDE(RE,P(n),x);
```

Warning: parameters have the values, $\{d=-4\,c, c=c, \alpha=\frac{1}{4}, b=2\,c, e=0, a=0\}$

$$\left[\frac{1}{2}(2\,x+1)\left(\frac{\partial^2}{\partial x^2}P(n,x)\right) - 2\,x\left(\frac{\partial}{\partial x}P(n,x)\right) + 2\,n\,P(n,x) = 0, \left[I = \left[\frac{-1}{2}, \infty\right], \rho(x) = 2\,e^{(-2\,x)}\right], \frac{k_{n+1}}{k_n} = 1\right]$$

```
> REtodiscreteDE(RE,P(n),x);
```

Warning: parameters have the values, $\{f=f, \alpha=\frac{f^2-1}{4\,f^2}, g=g, a=0, e=-g\,d,$

$$b = -\frac{1}{2}f\,d - \frac{1}{2}d, c = -\frac{1}{4}f^2\,d + \frac{1}{4}d + \frac{1}{2}g\,d\,f + \frac{1}{2}g\,d, d=d\}$$

$$\left[\frac{1}{2}\frac{(f+2\,x\,f-1)\,\Delta(\mathrm{Nabla}(\mathrm{P}(n,x\,f+g),x),x)}{f}-\frac{2\,x\,\Delta(\mathrm{P}(n,x\,f+g),x)}{1+f}\right.$$

$$+\frac{2\,n\,\mathrm{P}(n,x\,f+g)}{(1+f)\,f}=0,$$

$$\left[\sigma(x)=\frac{f}{2}+x-\frac{1}{2}-g,\sigma(x)+\tau(x)=-\frac{(f-1)\,(-1+2\,g-f-2\,x)}{2\,(1+f)}\right],\rho(x)=\left(\frac{f-1}{1+f}\right)^{x},$$

$$\left.\frac{k_{n+1}}{k_n}=\frac{1}{f}\right]$$

```
> strict:=true;
```

$$strict := true$$

```
> REtodiscreteDE(RE,P(n),x);
Error, (in REtodiscreteDE) this recurrence equation has no classical
discrete orthogonal polynomial solutions
```

Second example

```
> RE:=P(n+2)-x*P(n+1)+alpha*q^n*(q^(n+1)-1)*P(n)=0;
```

$$RE := \mathrm{P}(2+n)-\mathrm{P}(1+n)\,x+\alpha\,q^{n}\,(q^{(1+n)}-1)\,\mathrm{P}(n)=0$$

```
> REtoqDE(RE,P(n),q,x);
```

Warning: parameters have the values, $\{a=-d\,q+d, c=-\alpha\,q\,d+\alpha\,d, b=0, e=0, d=d\}$

$$\left[(x^2+\alpha)\,\mathrm{Dq}\left(\mathrm{Dq}\left(\mathrm{P}(n,x),\frac{1}{q},x\right),q,x\right)-\frac{x\,\mathrm{Dq}(\mathrm{P}(n,x),q,x)}{-1+q}+\frac{q\,(-1+q^{n})\,\mathrm{P}(n,x)}{(-1+q)^2\,q^{n}}=0,\right.$$

$$\left.\frac{\rho(q\,x)}{\rho(x)}=\frac{\alpha}{q^2\,x^2+\alpha},\frac{k_{n+1}}{k_n}=1\right]$$

```
>
```